Optimizing Irrigation Water Use Economics for Drought Adaptation

Summer 2013

Ministry of Agriculture and Water Resources
Erbil, Iraq
August 13 - 20

Professor
Frank A. Ward
New Mexico State University USA

Assisted by
Dina A. Salman
New Mexico State University USA

Support from
Saud A. Amer
US Geological Survey

Optimizing Irrigation Water Use Economics for Drought Adaptation
Workshop Plan and Schedule
Summer 2013
August 13-20

WORKSHOP MATERIAL: All workshop material, including slides, model code, model results, and software will be provided. A good reference source for our GAMS software is the user's guide. It's titled GAMS, A User's Guide, clickable at http://www.gams.com/dd/docs/bigdocs/GAMSUsersGuide.pdf

OBJECTIVE: Develop and apply optimization models that reduce farm income losses in periods of water shortage.

GRADING: Each participant who finishes the workshop will receive a certificate of completion.

EXERCISES: We plan to have several exercises, mostly using the computer. Exercises serve several purposes: break up the lecture routine, stimulate thinking, provide social interaction, gain confidence with the computer, and try out new ideas.

PROJECT: Please think about the workshop after you finish it. Sometime in the next two months, we would like to hear your comments on ways to improve this workshop or future ones like them.

Draft Daily Schedule
9:30 AM- 3:00 PM
Tuesday August, 13 - Tuesday August, 20

Tu, Aug 13	Introduction, overview, workshop mission and plans (FAW present, DAS and TH translate and help participants)
W, Aug 14	Model 1: Allocate land and water between 2 crops in 1 province (hands on computer if desired)
Th, Aug 15	Model 2: Allocate land and water between 2 crops and 2 provinces (hands on computer)
F, Aug 16	No class
Sat, Aug 17	Model 3: Allocate water shortage between 2 crops and 2 provinces (hands on computer)
Sun, Aug 18	Models 4-5: Assess rules for sharing water shortages (model 4) and improve display of results (model 5) (hands on computer)
M, Aug 19	Model 6: Expand scale to 13 provinces, 9 crops, 3 water supplies, and 4 shortage sharing methods (hands on computer)
T, Aug 20	Comments and Review, Certificates, Future Plans

Lunches: We have a daily lunch scheduled from 12 noon - 1 PM. Instructors will buy.
Dinners on your own

Introduction

Optimizing ag water resource use in the Tigris-Euphrates Basin: Drought Adaptation Strategies

- Frank A. Ward (NMSU)
- Dina Salman (NMSU)
- Saud A. Amer (USGS)
  - Erbil, Iraq
- Summer 2013

1

Objectives

- Learn and apply principle of optimization as framework for improved ag water use decisions.
- Learn the basic structure of the General Algebraic Modeling System (GAMS) software package, to be provided by the instructors.
- Discover what we mean by the economic value of water in agriculture.
- Learn how to put to better use for decision-making data on crops, water, and economics.

Objectives (cont)

- Learn how to build, use, and interpret optimization models written in GAMS to improve the performance of water use in crop irrigation for local, provincial, and national needs.
- Discover how optimization models can inform better capacity to adapt to water shortages in the lower Tigris-Euphrates Basin.
- (Bridge) Show how to use results so that:
  - farmers can increase income and improve livelihoods.
  - You can measure economic impacts of water shortage sharing methods.

Road Map

- Iraq drought adaptation challenges
- Economic Value of Water
- Introduction to Optimization
- A simple example
- GAMS software
- 6 Optimization Models in GAMS
  - Data needs
  - Math used
  - Results and Interpretation
- Conclusions

Motivation for Optimization Models

Iraq National Development plan (2010)

Water Resources Management Section

http://www.iauiraq.org/documents/1159/ndp24th.pdf

5.11.2 Water Policies
Integrated Management

4. Applying principles of integrated management of water resources in coordination with all stakeholders, …p 74.

Water Policies, continued
Wise Use

7. … Make available information so that water users are aware of the importance of the wise use of water … p. 74

Water Policies
conservation, drought, and salinity tolerance

9. … expanding application of modern irrigation methods and encourage planting substitute crops that consume less water and resist salinity and drought.

P.74

Water Policies
Climate impacts and growing scarcity

13. Study climate change, and its effects on water supplies into Iraq, and the future living conditions in light of expected scarcity, p. 74.

Water Policies
Research on increased water use efficiency

14. … increase the standard of irrigation; reduce waste; … apply research results to new projects, make information available to water users. Establish research centers…p 74

Water Policies
Pricing water to avoid waste and sustain supplies for future generations

15. … setting a suitable price for water to maintain its wealth… p 75

Economic Value of Water in Ag

- Information on water's economic value enables informed choices on water
  - development
  - conservation
  - allocation
  - purification
  - protection
- when growing demands for all uses occur with
  - increased scarcity
  - climate variability

13

Background

- Conceptually correct and empirically accurate estimates of the economic value of water are essential for rational allocation of scarce water across
  - locations
  - uses
  - quality levels
  - quantity levels
  - time periods
- Is economic rationality a good thing?

14

Economic Value of Water in Irrigation

- The economic value of water in agriculture is defined as what a rational irrigator is willing to pay. How large a check the farmer is willing to write.
  - Assumes that you want to use water to increase its total economic benefits in agriculture.
  - Irrigation policy debates typically focus on proposed marginal changes to existing supplies, quantities, or rules for sharing shortages.

15

Useful examples: value of water

- The economic value of protecting a country's water from under- deliveries by an upstream country is what the country's water users are willing to pay for the greater reliability of water supply.
- The economic value of protecting a region's irrigated agriculture is the gain in farm income and reduced food prices that the protection produces.
- The economic value of changing the current system for sharing water shortages when drought occurs is the gain in farm income and reduced food prices made possible by the new shortage sharing system.

16

Which policies need information on economic value of water?

- Value of added water infrastructure
  - building
  - Restoring
- Value of changing water institutions
  - Redrafting water laws
  - Stream adjudications (NM)
  - Shortage sharing rules (Iraq)
- Value of changing water regulations
  - Groundwater pumping limits
  - Requiring meters

17

Economic Value of Water

- It's the willingness to pay for making a change in the status quo of water
  - Changes in quantity
  - Changes in quality
  - Changes in timing
  - Changes in location
- It's about valuing changes in one of those four dimensions of water use.
- Usually requires an economic model grounded in a physical reality

18

Economic value of diverting irrigation water at a lower v higher point in watershed (e.g., Iraq)

- The willingness to pay for altering the diversion point from an upstream to a downstream region.
  - Farm income gain can be high
  - Gain in food security can be high
  - But existing rules may prevent the change
- So there can be an economic value in changing the rules for sharing shortages.

19

Background: Some Ag Water Challenges

- Afghanistan: Identify efficient timing, sizing, and sequencing of water storage projects (especially new storage reservoirs)
- Iraq: Identify agricultural water sharing systems for minimizing costs of adapting to drought in Lower Tigris-Euphrates
- USA: Protecting Farm Income when droughts occur
- Egypt: Identify improvements in water management that increase the Nile's economic productivity.
- Uzbekistan, Tajikistan: Identify Welfare Improving Developments and Allocations of Water in the Basin
- Australia: protecting key ecological assets without losing farm land.

20



Data

- Cropping Patterns (land in production by crop and year)
  - Rice, wheat, cotton, sunflower, Maize, Barley, summer vegetables, winter vegetables, and other crops
- Ag Water Use (Lower Tigris-Euphrates)
- Meteorological
  - Temperature, wind, humidity, sunshine
- Economic
  - Prices
  - Yields
  - Costs
  - Water use (ET)
  - Net Revenue per Ha

Iraq Map



Iraq Map

Q: What are we doing to do with this data to find out how to raise the economic value of water for irrigation in Iraq when drought occurs?

Goal: Reduce economic losses

A: Optimization

Optimization

- An optimization problem consists of maximizing or minimizing an objective by systematically choosing input values from an allowed set of choices. Result of optimization:
  - Best choices.
  - Results of best choices (value of objective).
- There are many optimization software packages, we use GAMS (General Algebraic Modeling System).

27

Optimization
Mathematical Formulation

28

Maximize Farm Income, by finding the best amount of land assigned to each crop in each province (L_{ik}) subject to a constraint of limited water

$$Max: Farm\ Income = \sum_i \sum_k NB_{ik}$$
$$i = province$$
$$k = crop$$

$$NB_{ik} = [P_{ik} * Yield_{ik} - Cost_{ik}] * L_{ik}$$
$$W_{ik} = Bc_{ik} * L_{ik}$$
$$\sum_i \sum_k W_{ik} \le W_0$$

29

Example Farm Water Data

Water Available (million m3)	Price ($/ton)	Yield (tons/ha)	Cost ($/ha)	Income ($/ha) P * Y - C	Water use (1000 m3/ha) Bc
572	Cotton: 906	Cotton: 2.32	Cotton: 1286	((906*2.32)– 1286) = ??	Cotton: 18.0
	Wheat 225	Wheat 1.39	Wheat: 200	((225*1.39) – 200) = ??	Wheat: 11.9

30

Challenge: Use previous slide's data to optimize total farm income

- Goal: Allocate land among crops that optimizes (maximizes) farm income
- Method:
  - Experiment with 3 possible land allocations
  - For larger problems, GAMS software tries all possible land allocations, then picks the best (optimum)
  - Data source: 1 Iraqi province (Mousil)

31

Remember

$$Total\ Income = \sum_i \sum_k [P_{ik} * Yield_{ik} - Cost_{ik}] * L_{ik}$$

32

Try 3 combinations of land planted to 2 crops, while using available water: 572 million m^3

	Land (1000 ha)	**Water Use (million m^3)**	**Total Farm Income ($US Thousands)**
1	Cotton= 0 Wheat=48.06	572	Cotton: **0 * 815.92 = 0** Wheat: **48.06 * 112.75 = 5,419** Total = 5,419
2	Cotton= 12.0 Wheat= 29.91		Cotton: **12.0 * 815.92 = 9,811** Wheat: **29.21 * 112.75 =3,626** Total = 13,437
3	Cotton= 31.78 Wheat= 0		Cotton: **31.78 * 815.92 = 25,928** Wheat: **112.75 * 0 = 0** Total= 25,928

33

Do all 3 of these crop land allocations exactly use up total water we have?

C	Land (1000 ha)	Water Use per ha (1000 m^3)	Total water use (million m^3)
1	Cotton= 0 Wheat=48.06	Cotton: 18.0 Wheat: 11.9	(0 * 18.0)+ (48.06 * 11.9) = 572
2	Cotton= 12.0 Wheat= 29.91		(12.0 * 18.0)+(29.91* 11.9) = 572
3	Cotton= 31.78 Wheat= 0		(31.78 * 18.0) + (0 * 11.9) = 572

For these 2 crops and 572 million m^3 of water, do you see other land allocations that use up all the water supply?

- How many allocations of land to those 2 crops would exhaust 572 million m^3 water?
- Answer ________?
- Challenge: We want the best allocation of land among crops
- Key word: __________

Q: What is Best Use of Land for Maximizing Total Farm Income?

Land in production by Crop (1000 ha):

- Cotton = 31.78, Wheat = 0

Water Use by Crop (million m3):

- Cotton = 572, Wheat = 0

Income by Crop ($ 1000 US):

- Cotton = $US 25,928, Wheat = 0

Total Income: = $25,928 (in 1000s)

Can you help the farmer make more income with information on the data slide.
They have no computers, and need advice.

Exercise: Free cup of tea for anybody who can make more than $25,928 (in 1000s) with available river water.

5-10 minute exercise.

37

Break

38

Let's think about the economic value of water in agriculture

- xx

39

Let's think about more crops and more provinces

There are more than 2 crops
and more than 1 province

40

Simple v Complex Models

- Simple Model
  - 2 crops
  - 1 province
  - 1 water supply scenario
- More Complex Models
  - Several crops
  - Several provinces
  - Several water supply possibilities
  - Several possible rules for sharing shortages among provinces

41

Basin Inflows Used

- Ag Water Use from Tigris + Euphrates
- Equal to total observed water use in irrigated agriculture from a base year (2006)
- Iraq Data from Saud Amer, January 2012

42

Optimization Software

43

Optimization Software

- We use GAMS (General Algebraic Modeling System) to model optimization of farm land/water use

GAMS Structure

SETS
(crops, provinces, water supply scenarios, shortage sharing rules, … time, use)

DATA
(prices, costs, yields, water supply)

VARIABLES
(water use, land use, farm income)

EQUATIONS
(objective functions and constraints)

SOLVER
(LP and NLP)

44

GAMS

- Development sponsored by World Bank in 1970s
- Went private in 1980s
- More, faster, and better solvers (1980s – today)
- Why GAMS is useful: Expandability
  - Toy sized models expand to real world application (models 1 – 6)
  - Suited for river basin optimization
  - Model structure easy to formulate
  - When more detail needed, add: crops, reservoirs, tributaries, seasons, years, water uses

45

Policy Assessment Framework



46

Course Plan:
Mission - Find a water allocation that maximizes farm income with available water supplies

- We plan to cover 6 models in the days ahead. All are stimulating and interesting.
- All optimize water use in irrigated agriculture.
- All use information on the economic value of water in irrigated agriculture.
- We'll have many class exercises
- All models and results are based on Iraqi data secured by Dr. Saud Amer in 2012
- Discuss needed model improvements, future classes

47

Model 1 (done)
You allocate land among 2 crops

- 2 crops: Wheat and Cotton
- 1 province: Mousil
- 1 Water constraint: 572 million m^3
- 572 million m^3 = observed use summed over those crops for that province.
- Goal: find the allocation of water among crops that <u>maximizes total farm income</u>.

48

Model 2

You allocate water
among crops and provinces

- 2 provinces: Mousil and Basra.
- 2 Crops: Wheat and Cotton.
- Water supply: 800.31 Million m^3.
- Goal: Maximize total farm income

Model 3

You assess economic impact of drought on irrigation

- 2 provinces: Mousil and Basra.
- 2 Crops: Wheat and Cotton.
- 2 Water Supplies: Normal (~800) v Dry (~400)
- Constraint: historical upper bound on land in production by crop and province
- Goal: Maximize Total Farm Income

Model 4

Allows you to economically assess methods of sharing shortages during drought

- 2 provinces: Mousil and Basra
- 2 Crops: Wheat and Cotton
- Constraint: upper bound by crop and province
- 2 water supplies: “Normal” v “Dry”
- 2 shortage sharing methods: upstream priority vs. proportional sharing of shortages
- Automatic multiple solves (over water supply and water sharing rules) inside loops
- Goal: Maximize Total Farm Income

Model 5:

You simplify display of results

- 2 provinces: Mousil and Basra
- 2 Crops: Wheat and Cotton
- Constraint: upper bound by crop and province
- 2 water supplies: Normal v Dry
- 2 Water sharing: upstream priority vs. proportional
- Same as model 4, but exports results to spreadsheet. Improves reading and interpreting of complex results.
- Saves to spreadsheet shadow price (economic value of additional water) by water supply and water sharing rule
- Goal: Maximize Total Farm Income

52

Model 6: Monster

You take a national view of ag/water use
with large # of scenarios and water sharing rules

- 13 provinces
- 9 Crops
- Constraint: upper bound by crop and province
- 3 water supplies: Normal, Dry, Drought
- 4 Water shortage sharing rules: upstream priority, downstream priority, proportional sharing, free market trading
- Export results to Excel sheet for display.
- Calculate shadow prices: economic value in ag of added water ($US/1000 cubic meters) if the added water could be found, developed, pumped, conserved, or negotiated.
- Goal: Maximize Total Farm Income

53

Model 6: Monster

Lets you analyze how losses in farm income at the farm and basin levels is affected by various water allocation methods for adapting to drought of various severity.

54

Conclusions

55

Water Economics Optimization Models

- Comparisons across country/basin/region/project
- Quantifies incentives to meet social aims
- Consistent calculation npv of water programs
- Can be used to guide policy reform (e.g. regulations, pricing, or water sharing rules)
- Several layers of detail targeted by different audiences

Water Economics Optimization Models

- Can be applied to any country/basin/region/project
- Flexibility of inputs and outputs
- Capacity for sensitivity analysis ($\partial Y_i/\partial a_j$)
- Values water in alternative uses, time periods, crops, locations

Top 10 Lies told by Water Modelers

1. The model is well-documented with all limits
2. The model is user-friendly
3. The model fits the data
4. Results make sense
5. The model does that
6. We did a sensitivity analysis
7. Anyone can run this model
8. This model links to other models
9. The model will be in the public domain
10. The new version fixes all previous problems

58

Model 1

Code

```
$ EOLCOM //
$ TITLE Optimizing Water Resource Use in the TE Basin - Iraq
$ OFFLISTING

OPTIONS LIMROW = 0, LIMCOL = 0

$ONTEXT
EOLCOM // tells GAMS to ignores everything in a line after the symbol //
OFFLISTING deletes all program lines and just includes GAMS results saves paper

Set LIMROW = 0 to eliminate all equations in the GAMS listing
Set LIMROW = 100 or more to show all equations in listing. Can help trap errors
Set LIMCOL = 0 to eliminate starting values of all variables (rarely needed).

* ---------------------------------------------------------------------------------------

  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  Dina A. Salman:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, New Mexico USA
  email: dinasalm@nmsu.edu

  Saud A.Amer:
  US Geological Survey - International Water Resources Division
  Reston VA  USA
  email:  samer@usgs.gov

  May 13, 2013

* ---------------------------------------------------------------------------------------
*  Model 1: Simple Farm Management: optimizes (maximizes) total farm income
*  2 crops and 1 province
*  Limited water,  Unlimited land
* ---------------------------------------------------------------------------------------
$OFFTEXT
```

```
****************************************   Section 1 ***********************************
*                                           Sets                                       *
****************************************************************************************

SETS

k  crop

/  1-Wheat
   2-Cotton
/

***************************  Section 2 *************************************************************************
*                               Data                                                                           *
****************************************************************************************************************

parameter  Bc(k)    crop water demand (ET)    (1000s cubic meters per Ha = 10ths of meters depth)

/1-wheat   11.9
 2-cotton  18.0/

*Bc(k) =  0.5 * Bc(k)

parameter    Yield_p(k) Crop Yield - often proportional to ET    (tons per Ha)

/1-wheat    1.39
 2-cotton   2.32
/

* economic data

Parameter   Price_p(k)  Crop Prices                              ($ US per ton)

/1-Wheat    225
 2-Cotton   906
/

parameter Cost_p(k)   Production costs excluding water           ($US per Ha)

/1-wheat    200
```

```
 2-cotton  1286/

parameter Wat_supply_p   total water available                   (million cubic meters per year)

/572/

Parameter Net_revenue_p(k)  calculated net revenue per unit land ($ US per Ha)
;

Net_revenue_p(k) = Price_p(k) * Yield_p(k) - Cost_p(k);

***********************************************  Section 3 ***********************************************
*                                                Variables                                                *
***********************************************************************************************************
positive variables

hectares_v   (k)      land in production by crop       (1000 Ha - marginal is $US per Ha)
T_hectares_v          total land in production         (1000 Ha - marginal is $US per Ha)
uses_crop_v  (k)      total water use by crop          (million m^3 - marginal is $US per 1000 m^3)
uses_v                total water use                  (million m^3 - marginal is $US per 1000 m^3)

variables
ag_ben_k_v   (k)      total farm income by crop        ($US 1000s - no marginals shown)
ag_ben_v              summed farm income (objective)    ($US 1000s - no marginals shown)

**********************************************   Section 4  **********************************************
*                                                Equations                                               *
**********************************************************************************************************

// equations declared (named)

Equations

T_hectares_e          total land in production         (1000 Ha -  marginal is $US per Ha)
Uses_crop_e  (k)      total water use by crop          (million m^3 - marginal is $US per 1000 m^3)
uses_e                total water use                  (million m^3 - marginal is $US per 1000 m^3)

ag_ben_k_e   (k)      total farm income by crop        ($US 1000s - no marginals shown)
ag_ben_e              total farm income (objective)    ($US 1000s - no marginals shown)
;
```

```
// equations defined (uses declared names to write with algebraic formulas)

T_hectares_e   ..   T_hectares_v     =e=  sum(k,   hectares_v(k));          // sums land in prodn over crops
Uses_crop_e (k)..   uses_crop_v (k)  =e=   Bc(k) * hectares_v(k);           // total water use by crop
uses_e         ..   uses_v           =e=  sum(k,  uses_crop_v(k));          // total water use summed over crops
ag_ben_k_e  (k)..   ag_ben_k_v  (k)  =e=  net_revenue_p(k) * hectares_v(k); // total farm income by crop
ag_ben_e       ..   ag_ben_v         =e=  sum(k,   ag_ben_k_v(k));          // objective fn:  total farm income

*************************************** Section 5 ****************************************************************
*  labels and defines all models used                                                                            *
*  Each model has one objective
******************************************************************************************************************

model TE_01 /all/;                                                 // uses all equations above

**************************************  Section 6   **************************************************************
*                                        Bounds                                                                  *
*  Setting bounds helps you discover shadow price (economic value) for each limiting resource                    *
******************************************************************************************************************

uses_v.up = wat_supply_p;          // upper bound on total water availablem, avoids unbounded solution
*uses_crop_v.lo('wheat')  =  100;  // lower bound on water assigned to wheat

************************************** Section 7 *****************************************************************
*                                    SOLVE THE MODEL                                                             *
******************************************************************************************************************

solve TE_01 using nlp maximizing ag_ben_v;  // uses nonlinear programming solver

************************************ THE END *********************************************************************
```

Model 1

Listing

Model 1
Results Listing and Comments

```
GAMS Rev 233 WIN-VIS 23.3.3 x86/MS Windows
05/15/13 13:58:43 Page 3
Optimizing Water Resource Use in the TE Basin - Iraq
Solution Report     SOLVE TE_01 Using NLP From line 153


               S O L V E      S U M M A R Y

     MODEL   TE_01                OBJECTIVE  ag_ben_v   Name of variable used for objective
     TYPE    NLP                  DIRECTION  MAXIMIZE   Maximize don’t minimizes
     SOLVER  CONOPT               FROM LINE  153

**** SOLVER STATUS     1 Normal Completion    Normal is good – anything else is bad
**** MODEL STATUS      1 Optimal              Optimal is good - infeasible or unbounded is bad
**** OBJECTIVE VALUE            25928.1244

 RESOURCE USAGE, LIMIT          0.008      1000.000     Good
 ITERATION COUNT, LIMIT         4    2000000000         Good
 EVALUATION ERRORS              0             0         Good
```

Click onto Sol_VAR then to hectares_v on your computer's GAMS listing

```
---- VAR hectares_v  land in production by crop        (1000 Ha - marginal is $US per Ha)

                LOWER          LEVEL          UPPER         MARGINAL

1-Wheat           .              .            +INF             .
2-Cotton          .            31.7778        +INF             .
```

Level: LEVEL tell you the model's optimized quantities for the variable hectares_v. These are the hectares of land in production by crop. The number tells you that farmers maximize their total income for Mousil Province by planting _____________ ha to cotton, and ______________ha to wheat.

LOWER equals . for both crops. That means lower bound is 0 for both wheat and cotton. Of course, land in production can never be negative. UPPPER is +INF for both crops. That means the upper bound is infinite for both crops. So LEVEL of hectares_v for both crops must be between 0 and infinity.

Table tells you the net income-maximizing combination of land planted to each crop. Remember, you have only 572 million cubic meters of water available total for both crops.

The table advises the farmer to plant all cotton and no wheat. Why ______________?

But wheat, you say, is an important grain crop. Why shouldn't farmers plant wheat ____________?

Later we can try re-running the model by reducing cotton and increasing wheat, and see what happens. You'll find it reduces total farm income.

Marginal: Table tells you MARGINAL impact of increasing the optimized hectares of land by one unit. Both MARGINALS are zero. That tells you added net income from one more unit of land brought into production. What does the zero mean _________________? More land has no value without more water.

NOTES:

Click onto T_hectares_v

```
                              LOWER          LEVEL          UPPER         MARGINAL

---- VAR T_hectare~             .           31.7778         +INF              .

  T_hectares_v  total land in prodn                 (1000 Ha - marginal is $US per Ha)
```

Levels: Table LEVELS tell you optimized LEVEL of the variable T_hectares_v, which is total land in production, summed over crops. It tells you that ________ is optimized total land in production.

The LEVEL 31.78 tells you total land to plant to both crops with available surface water.

Does the LEVEL of the variable T_hectares_v measure total income ___________?
Does the LEVEL of the variable T_hectares_v measure total water use __________?
Does the LEVEL of the variable T_hectares_v measure land planted to any one crop _________?
Does the LEVEL of the variable T_hectares_v measure total land planted to all crops ________?

Bounds: LOWER and UPPER bounds are still the same, 0 and infinity

Marginals: Notice that the MARGINAL is ______________ What does that mean ____________?

NOTES

GAMS Rev 233 WIN-VIS 23.3.3 x86/MS Windows
05/15/13 13:58:43 Page 5
Optimizing Water Resource Use in the TE Basin - Iraq
Solution Report   SOLVE TE_01 Using NLP From line 153

---- VAR uses_crop_v total water use by crop   (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
1-Wheat	.	.	+INF	-35.8541
2-Cotton	.	572.0000	+INF	.

Levels: This result tells you the optimized LEVEL of total water use for each crop, uses_crop_v. The optimized LEVEL of total water use is ___________ for wheat and ___________ for cotton.

The table tells you that all 572 million cubic meters of water should be used to grow cotton.

Marginals: They tell you the MARGINAL impact (added value) of adding one unit of water to either crop.

Notice the negative MARGINAL for wheat = -35.85. What does that mean _________? It tells the farmer or water manager that if Iraqi farmers insist on growing wheat for any reason, there is a loss $US 35.85 for the first extra unit (1000 cubic meters) of water taken from cotton and planted to wheat.

Why is there a loss of $35.85? Does that mean that wheat is an absolute money loser __________?

NOTES:

```
                                LOWER          LEVEL          UPPER         MARGINAL

---- VAR uses_v                   .          572.0000       572.0000        45.3289

  uses_v total water use           (million m^3 - marginal is $US per 1000 m^3)
```

Levels: table tells you optimized LEVEL of uses_v. That variable is total water use applied to all land. The optimized LEVEL is _____________? (in million cubic meters). The model recommends using all total water available (572) on crop irrigation. None is wasted.

Upper Bound: The UPPER BOUND is ____________. So notice that the optimized LEVEL of total water use is also equal to its UPPER bound. Interesting. When any resource enters the optimal solution LEVEL at its UPPER BOUND, the results tell you that the model is hungry for more.

Marginals: Notice the MARGINAL equals $US 45.33. This gives you important information for future planning. It tells you if farmers could find one more unit of water (in thousand cubic meters) from any source, they earn $US 45.33 more net income. What are examples of new water sources? rain, conservation, infrastructure, purchase, rent, ditch lining, new storage capacity, reduced pump costs.

Q: Suppose you're skeptical. You don't believe the MARGINAL = $US 45.33. How can you check to see if 1 more (thousand cubic meters) is worth to the farmer $US 45.33 ______________?

It also tells you the loss in income in $US if you had 1 less (thousand cubic meters) of water.

Q: How can you check to see if 1 unit less water available reduces your income by $US 45.33 ________?

NOTES

---- VAR ag_ben_k_v total farm income by crop ($US 1000s - no marginals shown)

	LOWER	LEVEL	UPPER	MARGINAL
1-Wheat	-INF	.	+INF	.
2-Cotton	-INF	25928.1244	+INF	.

Level: This result shows the optimal LEVEL of total farm income by crop (ag_ben_k_v). That optimal LEVEL is __________________?

It does not tell you what farmers actually earned. It tells you what farmers could earn if they want to maximize their income for the information given. What information did we give the model ________________? The model has information on prices, yields, costs, water use, and total water available.

The table tells Iraqi farmers and extension advisers in Mousil Province to grow only cotton, and to plant no wheat at all if income maximization is their goal.

Compared to cotton, planting wheat loses money. Wheat earns some income, but less than cotton. That is, cotton earns more income than wheat. Based on the data we found on prices and costs, maximum net income that could be earned from these two crops (wheat and cotton) is $ US 25,928 (in thousands).

This tells you what helps the farmers. However, suppose the government wants farmers to grow wheat to support national grain self-sufficiency __________________?

Marginal: No marginal shown for the variable total farm income. Income is the objective. It is not a potentially scarce resource like land or water.

NOTES

```
                         LOWER          LEVEL          UPPER         MARGINAL

---- VAR ag_ben_v        -INF        25928.1244        +INF              .

  ag_ben_v  total farm income (objective)      ($US 1000s - no marginals shown)
```

Level: This table shows ag_ben_v. It is the optimized LEVEL of total farm income that can be earned from both crops if all available water is put to best use in irrigated agriculture. That LEVEL is ___________. If farmers don't use their water wisely, they won't earn this much income. It's the optimized LEVEL of income, not necessarily the actual level.

Notice that the table shows no breakdown between cotton and wheat. It shows only total net income that could be earned by irrigation farmers from the province. It exactly matches the optimal solution shown on page 1.

What if the government wanted to help farmers raise their income beyond this level __________?

Marginals: None shown because LOWER bound is – infinity and UPPER bound is + infinity.

NOTES

Model 2

Code

```
$EOLCOM //
$TITLE Optimizing Water Resource Use in the TE Basin - Iraq
$OFFLISTING OFFUPPER

OPTION LIMROW = 0, LIMCOL = 0;

$ONTEXT

* ------------------------------------------------------------------------------------
  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  Dina A. Salman:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, New Mexico USA
  email: dinasalm@nmsu.edu

  Saud A.Amer:
  US Geological Survey - International Water Resources Division
  Reston VA  USA
  email:  samer@usgs.gov

  May 13, 2013

* ------------------------------------------------------------------------------------
*  Model 2 Simple Farm Management
*  Adds a SET for province (2 provinces: Mousil-Basra)
*  Expands water supply to 800 million cubic meters (2006 use level for both 2 provinces)
* ------------------------------------------------------------------------------------

$OFFTEXT

****************************************   Section 1 ***********************************
*                                            Sets                                      *
***************************************************************************************
```

```
SETS

i    province

/
  1-Mousil
  2-Basra
/

k    crop

/  1-wheat
   2-cotton
/


************************************  Section 2 ******************************************
*                                         Data                                           *
******************************************************************************************

table  Bc(i,k)    crop water demand (ET) (1000s cubic meters per Ha = 10ths meters depth)

                 1-wheat          2-cotton
 1-mousil          11.9              18.0
 2-Basra           13.5              21.4

*Bc(i,k) =  0.5 * Bc(i,k)

table    Yield_p(i,k)   Crop Yield - proportional to ET (tons per Ha)

               1-wheat         2-Cotton
1-Mousil        1.39              2.32
2-Basra         1.57              2.75


* economic data

parameter   Price_p(k)   Crop Prices ($ US per ton)

/1-Wheat        225
```

```
 2-Cotton       906
/

table Cost_p(i,k)     Production Costs Excluding water ($US per Ha)

                1-wheat     2-cotton
1-Mousil          200        1286
2-Basra           207        1345

parameter Wat_supply_p     total water available (million cubic meters per year)

/800/

Parameter Net_revenue_p(i,k)  calculated net revenue per unit land ($ US per Ha)
;

Net_revenue_p(i,k) = Price_p(k) * Yield_p(i,k) - Cost_p(i,k);

******************************  SECTION 3 *********************************************
*                               Variables                                             *
**************************************************************************************
positive variables

hectares_v    (i,k)   land in production by crop-province   (1000 Ha - marginal is $US per Ha)
T_hectares_v  (i  )   total land in prodn by province       (1000 Ha - marginal is $US per Ha)
uses_crop_v   (i,k)   total water use by crop-province      (million m^3 - marginal is $US per 1000 m^3)
uses_v                total water use                       (million m^3 - marginal is $US per 1000 m^3)

variables
ag_ben_k_v    (i,k)   total farm income by crop             ($US 1000s - no marginals shown)
ag_ben_v              total farm income (objective)         ($US 1000s - no marginals shown)

*******************************   Section 4  *******************************************
*                                 Equations                                           *
**************************************************************************************
Equations

// equations declared (named)

T_hectares_e   (i  )  total land in production by province   (1000 Ha - marginal is $US per Ha)
```

```
Uses_crop_e    (i,k)  total water use by crop - province     (million m^3 - marginal is $US per 1000 m^3)
uses_e                total water use                        (million m^3 - marginal is $US per 1000 m^3)

ag_ben_k_e     (i,k)  total farm income by crop - province   ($US 1000s - no marginals shown)
ag_ben_e              total farm income (objective)          ($US 1000s - no marginals shown)
;

// equations defined (using above names with algebraic formulas)

T_hectares_e (i  ).. T_hectares_v (i  )  =e=  sum(k, hectares_v     (i,k));          // sums land in prodn over crops
Uses_crop_e  (i,k).. uses_crop_v  (i,k)  =e=  Bc(i,k) * hectares_v  (i,k);           // total water use by crop-province
uses_e            .. uses_v              =e=  sum((i,k), uses_crop_v(i,k));         // total water use summed over crops-prov
ag_ben_k_e   (i,k).. ag_ben_k_v   (i,k)  =e=  net_revenue_p(i,k) * hectares_v(i,k);  // total farm income by crop-province
ag_ben_e          .. ag_ben_v            =e=  sum((i,k), ag_ben_k_v (i,k));         // objective fn total farm income

*************** Section 5 ***************************************************************
*  Labels and defines all models used                                                   *
*  Each model has one objective                                                         *
*****************************************************************************************

model TE_02 /all/ ;                                                // uses all equations above

**********************************  Section 6   ****************************************
*                                   BOUNDS                                              *
*  Bounding shows a positive shadow price for each limiting resource                    *
*****************************************************************************************

uses_v.up = wat_supply_p;   // upper bound on total water available, avoids unbounded solution
*hectares_v.lo('1-mousil','1-wheat') =  10; // food security requirement
*uses_crop_v.lo('wheat')  = 100;            // lower bound on water assigned to wheat

**************** Section 7 **************************************************************
*                SOLVE THE MODEL                                                        *
*****************************************************************************************

solve TE_02 using nlp maximizing ag_ben_v;  // uses nonlinear programming solver

**************** Section 7 **************************************************************
*                THE END                                                                *
*****************************************************************************************
```

Model 2

Listing

Model 2
Results Listing and Comments

```
GAMS Rev 233  WIN-VIS 23.3.3 x86/MS Windows
05/15/13 16:57:37 Page 3
Optimizing Water Resource Use in the TE Basin - Iraq
Solution Report     SOLVE TE_02 Using NLP From line 167


               S O L V E      S U M M A R Y

     MODEL   TE_02                OBJECTIVE  ag_ben_v  name of objective variable
     TYPE    NLP                  DIRECTION  MAXIMIZE  direction of optimization
     SOLVER  CONOPT               FROM LINE  167

**** SOLVER STATUS     1 Normal Completion         good
**** MODEL STATUS      1 Optimal                   good, not Infeasible, not Unbounded
**** OBJECTIVE VALUE            42859.8131

 RESOURCE USAGE, LIMIT          0.004      1000.000   good
 ITERATION COUNT, LIMIT         4    2000000000       good
 EVALUATION ERRORS              0             0       good
```

GAMS Rev 233 WIN-VIS 23.3.3 x86/MS Windows
05/15/13 16:57:37 Page 5
Optimizing Water Resource Use in the TE Basin - Iraq
Solution Report SOLVE TE_02 Using NLP From line 167

---- VAR hectares_v land in production by crop-province (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
1-Mousil.1-wheat	.	.	+INF	.
1-Mousil.2-cotton	.	.	+INF	.
2-Basra .1-wheat	.	.	+INF	.
2-Basra .2-cotton	.	37.3832	+INF	.

Levels: Table remind us that we added a new province, Basra. The optimized LEVEL of land assigned to wheat is zero for both provinces. The LEVEL of land for cotton is ______ for Mousil, and _______ (1000 ha) for Basra.

By adding new opportunities, use of water for crops in Basra, the model takes advantage of the opportunity.

If optimizing total income is the goal, farmers should grow only cotton and only cotton in Basra. Why should they grow no wheat ___________? Why should they grow nothing in Mousil ___________?

Notice the optimized LEVEL of land, 37.38 (1000 ha), is larger than the 31.78 (1000 ha) used to grow cotton in in model 1. Why is land larger in model 2 than in model 1 ______________________? Did we tell the model that we now have more land ___________________?

Impact of more water: For model 2, we gave our farmers more water, 800 million cubic meters for both provinces. Model 1 gave farmers only 572 million cubic meters. More water means more land will be irrigated.

An optimization model always wants to optimize something (e.g. income). It's hungry for income. An optimization model looks for the allocation of water and land among crops and provinces to maximize income. Basra has higher cotton yields than Mousil (hotter climate), so higher revenue but only slightly higher costs.

Suppose Iraq's national water rights system assigns half (400) to crops in Basra, and half (400) to crops in Mousil (400). What will happen _________? Can you find a way to raise income given those rules _________?

Marginals: What do the 4 MARGINALS equal to zero mean ____________? Added land, if it could be found, is worth nothing. Is that really true __________? Added land without added water is worth nothing.

NOTES

```
---- VAR T_hectares_v total land in prodn by province       (1000 Ha - marginal is $US per Ha)

                LOWER          LEVEL          UPPER         MARGINAL

1-Mousil          .              .            +INF              .
2-Basra           .            37.3832        +INF              .
```

Levels: Table shows total optimized LEVEL of land in production summed over crops, for each province, Mousil and Basra, T_hectares_v.

It shows you what you'd expect from the previous table. 37.38 (1000 ha) is total land that would be in production if income maximization is the goal.

Is that really the goal for the farmer ___________?

Is it or should it be the goal for the country ______________?

What are the limits of income maximization models, given that this one gives nothing to Mousil ___?

Marginals: Both MARGINALS are _____________. What do those two zeros mean ______________?

NOTES

```
---- VAR uses_crop_v total water use by crop-province    (million m^3 - marginal is $US per 1000 m^3)

                         LOWER          LEVEL          UPPER         MARGINAL

1-Mousil.1-wheat           .              .            +INF          -44.1000
1-Mousil.2-cotton          .              .            +INF           -8.2459
2-Basra .1-wheat           .              .            +INF          -42.7414
2-Basra .2-cotton          .           800.0000        +INF             .
```

Levels: This table summarizes the optimized LEVEL of uses_crop_v. Optimized level of each crop in each province is what ________________.

It tells us that all 800 (million cubic meters) water should be applied to cotton and only in Basra, if income optimization is the goal.

The model advises complete specialization of water for what crop __________ in what province _________?

Marginals: The MARGINALS give you valuable planning information that is important to inform water policy. What information do they tell you ___________?

MARGINALS tell you that if Iraqi farmers insist on applying 1 more (million cubic meters) in Mousil, there will be loss 44.10 ($US) on the 1st 1000 cubic meters for wheat and 8.24 ($US) on the 1st 1000 cubic meters for cotton.

There’s also a loss of $US 42.74 for 1 extra (1000 cubic meters) water applied to wheat in Basra. Does that mean that you should not grow wheat in Basra ___________________?

If you insist on putting water to another crop or to another province you will suffer a cost in lost farm income.

NOTES

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR uses_v	**.**	**800.0000**	**800.0000**	**53.5748**

uses_v total water use (million m^3 - marginal is $US per 1000 m^3)

Levels: This table tells you the LEVELS of uses_v, total water summed over crops and provinces.

What is that LEVEL ____________________?

It says that you should use all of your available water and let none go to waste. Recall that we told the GAMS model that the UPPER bound on water is 800 (million cubic meters).

Marginal: The MARGINAL (53.57) tells you lots of things:

1. Additional water is valuable and will be used for crop irrigation.
2. It could be a good idea to find or develop more water. How could this happen _______? desalination, treaties with upstream neighbors, conserving water in irrigation (difficult), developing drought resistant crops, reduced urban water use, reduced flows for key ecological assets, reduced use in urban areas, reduced outflows at Shatt El Arab to the Gulf.
3. It tells you the maximum you can justify spending to get the additional water. How much ________?
4. It says that Iraqi farmers can gain $US 53.47 in added farm income per 1000 cubic meters added water if somebody could help them find added water

NOTES

```
---- VAR ag_ben_k_v  total farm income by crop          ($US 1000s - no marginals shown)

                         LOWER          LEVEL          UPPER         MARGINAL

1-Mousil.1-wheat         -INF             .            +INF             .
1-Mousil.2-cotton        -INF             .            +INF             .
2-Basra .1-wheat         -INF             .            +INF             .
2-Basra .2-cotton        -INF        42859.8131        +INF             .
```

This table tells us the optimized LEVEL of ag_ben_k_v, total farm income by crop. It repeats what we already know about total optimized farm income. How much income should be earned by each crop it total income is what should be optimized ______________________?

It advises Iraqi farmers that it's best (optimized) to earn income by farming only cotton and farming only in Basra. If farmers and ministry officials take the advice shown in the table, they'll earn 42,859 (in $US 1000s).

Why are MARGINALS zero ___________________?

NOTES

```
GAMS Rev 233  WIN-VIS 23.3.3 x86/MS Windows
05/15/13 16:57:37 Page 6
Optimizing Water Resource Use in the TE Basin - Iraq
Solution Report     SOLVE TE_02 Using NLP From line 167

                           LOWER          LEVEL          UPPER         MARGINAL

---- VAR ag_ben_v          -INF        42859.8131        +INF             .

  ag_ben_v  total farm income (objective)         ($US 1000s - no marginals shown)
```

The table shows the optimized LEVEL OF ag_ben_v, namely the objective function (total farm income).

What is its interpretation ____________? It’s the maximum total net income that can be earned from all (both) crops in all (both) provinces.

But farmers must pay careful attention to their revenues and costs to earn this much. They cannot make mistakes. It’s important for farmers and farm advisers to have good data, and to use it well.

NOTES

Model 3

Code

```
$EOLCOM //
$TITLE Optimizing Water Resource Use in the TE Basin - Iraq
$OFFSYMXREF OFFSYMLIST OffLISTING OFFUPPER

OPTION LIMROW=000, LIMCOL = 0;

$ONTEXT

* ---------------------------------------------------------------------------------------

  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  Dina A. Salman:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, New Mexico USA
  email: dinasalm@nmsu.edu

  Saud A.Amer:
  US Geological Survey - International Water Resources Division
  Reston VA  USA
  email:  samer@usgs.gov

  May 13, 2013

* ---------------------------------------------------------------------------------------
*  Model 3 Simple Farm Management
*
*  Adds a SET for water supply scenario (2 scenarios: normal-dry)
*  Adds data on observed land in production (Thanks to Saud Amer)
*  Normal water use calculated parameter: observed land in prodn multiplied by crop water use
* ---------------------------------------------------------------------------------------

$OFFTEXT
```

```
**************************************** Section 1 ***********************************
*                                          Sets                                      *
**************************************************************************************

SETS

i    province

/
  1-Mousil
  2-Basra
/

k    crop

/  1-wheat
   2-cotton
/

s   hydrologic water supply scenario

/
normal
dry

/

************************************ Section 2 ***************************************
*                                      Data                                          *
**************************************************************************************

table  Bc(i, k)    crop water demand (ET) (1000s cubic meters per Ha = 10ths meters depth)

                 1-wheat          2-cotton
 1-mousil          11.9              18.0
 2-Basra           13.5              21.4

*Bc(k) =  0.5 * Bc(k)

table    Yield_p(i,k)   Crop Yield tons per HA (proportional to ET)

               1-wheat         2-Cotton
```

```
1-Mousil        1.39             2.32
2-Basra         1.57             2.75


* economic data


parameter   Price_p(k)        Crop Prices ($ US per ton)

/1-Wheat        225
 2-Cotton       906
/

table Cost_p(i,k)     Production Costs excluding water ($US per Ha)

                1-wheat   2-cotton
1-Mousil          200      1286
2-Basra           207      1345

;

Table land_p(i,k)   observed land in prodn from year 2006 (1000 Ha)

                 1-wheat        2-cotton
1-Mousil           47.4           0.45
2-Basra            16.9           0.00
;

Parameter Net_revenue_p(i,k)  calculated net revenue per unit land ($ US per Ha)
;

Net_revenue_p(i,k) = Price_p(k) * Yield_p(i,k) - Cost_p(i,k);


parameter Wat_supply_p(s)   total water used: full supply = year 2006 (million cubic meters per year)
;
Wat_supply_p('normal') = sum((i,k), land_p(i,k) * Bc(i,k));   // calculates observed water use
Wat_supply_p('dry')    = 0.5 * wat_supply_p('normal');        // dry: half of observed water use

******************************  SECTION 3 *********************************************
*                               Variables                                            *
**************************************************************************************
```

```
positive variables

hectares_v     (i,k,s)   land in production by crop-province-scen    (1000 Ha - marginal is $US per Ha)
T_hectares_v   (i,   s)   total land in prodn by province-scen        (1000 Ha - marginal is $US per Ha)
uses_crop_v    (i,k,s)   total water use by crop-province-scen       (million m^3 - marginal is $US per 1000 m^3)
uses_v         (    s)   total water use by scen                     (million m^3 - marginal is $US per 1000 m^3)

variables

ag_ben_k_v     (i,k,s)   total farm income by crop-province-scen     ($US 1000s - no marginals shown)
ag_ben_v       (    s)   total farm income by scen                   ($US 1000s - no marginals shown)

ag_ben_normal_v          total farm income (objective - full water)    ($US 1000s - no marginals shown)
ag_ben_dry_v             total farm income (objective - reduced water) ($US 1000s - no marginals shown)


*******************************   Section 4  *********************************************
*                                  Equations                                               *
*****************************************************************************************

Equations

//equations declared

T_hectares_e   (i,   s)    total land in production by province-scen  (1000 Ha - marginal is $US per Ha)
Uses_crop_e    (i,k,s)    total water use by crop-province-scean     (million m^3 - marginal is $US per 1000 m^3)
uses_e         (    s)    total water use by scen                    (million m^3 - marginal is $US per 1000 m^3)

ag_ben_k_e     (i,k,s)    total farm income by crop-province-scen    ($US 1000s - no marginals shown)
ag_ben_e       (    s)    total farm income by scen                  ($US 1000s - no marginals shown)

ag_ben_normal_e           total farm income (obj 1-full water)       ($US 1000s - no marginals shown)
ag_ben_dry_e              total farm income (obj 2-reduced water)    ($US 1000s - no marginals shown)
;

// equations defined (using above names with algebraic formulas)

T_hectares_e   (i,   s)..  T_hectares_v(i,  s)   =e=  sum(k,   hectares_v   (i,k,s));   // land in prodn by prov-scen
Uses_crop_e    (i,k,s)..  uses_crop_v (i,k,s)   =e=  Bc(i,k) *  hectares_v (i,k,s);    // water use by crop-prov-scen
uses_e         (    s)..  uses_v      (    s)   =e=  sum((i,k),uses_crop_v (i,k,s));        // total water use by scen
ag_ben_k_e     (i,k,s)..  ag_ben_k_v  (i,k,s)   =e=  net_revenue_p(i,k) * hectares_v(i,k,s); // income crop-prov-scen
```

```
ag_ben_e        (    s).. ag_ben_v    (    s)   =e=  sum((i,k),  ag_ben_k_v(i,k,s));  // total farm income by scen
ag_ben_normal_e        .. ag_ben_normal_v       =e=  ag_ben_v('normal');      // obj fn 1: income full water scen
ag_ben_dry_e           .. ag_ben_dry_v          =e=  ag_ben_v(   'dry');      // obj fn 2: income reduced water scen

*************** Section 5 ***************************************************************
*  Labels and defines all models used                                                    *
*  Each model has one objective function                                                        *
******************************************************************************************

model TE_03 /all/ ;

**********************************  Section 6   ******************************************
*                                   BOUNDS                                               *
*  Bounding shows a positive shadow price for each limiting resource                     *
******************************************************************************************

uses_v.up    (    s) = wat_supply_p(s);  // upper bound on total water available (2 scenarios)
hectares_v.up(i,k,s) = land_p(i,k);      // can allocate no more land to crops than full water (2006)

*hectares_v.lo('1-mousil','1-wheat') =  10; // food security requirement
*uses_crop_v.lo('wheat')  = 100;            // lower bound on wter assigned to wheat

**************** Section 7 ***************************************************************
*               SOLVE THE MODEL                                                          *
******************************************************************************************

* model 1
solve TE_03 using nlp maximizing ag_ben_normal_v;

* model 2
uses_crop_v.fx(i,k,'normal') = uses_crop_v.l(i,k,'normal');  // save optimized water use from 1st model solution

solve TE_03 using nlp maximizing ag_ben_dry_v;

// THE END
```

Model 3

Listing

MODEL 3: ADDS 2 WATER SUPPLY SCENARIOS

- **OPTIMIZES FARM INCOME FOR FULL AND REDUCED WATER SUPPLY**
- **UPPER BOUND ON LAND IN PRODUCTION BY CROP = 2006 LEVELS**

2 MODELS TOTAL

3.1: FULL (100% OF 2006) WATER SUPPLY

```
GAMS Rev 233  WIN-VIS 23.3.3 x86/MS Windows
05/15/13 18:46:36 Page 3
Optimizing Water Resource Use in the TE Basin - Iraq
Solution Report     SOLVE TE_03 Using NLP From line 197


               S O L V E      S U M M A R Y

     MODEL   TE_03               OBJECTIVE  ag_ben_normal_v
     TYPE    NLP                 DIRECTION  MAXIMIZE
     SOLVER  CONOPT              FROM LINE  197

**** SOLVER STATUS     1 Normal Completion      good
**** MODEL STATUS      1 Optimal                good
**** OBJECTIVE VALUE             8183.1390      good (lower income because of upper bound land constraint)

 RESOURCE USAGE, LIMIT          0.008      1000.000    good
 ITERATION COUNT, LIMIT         5    2000000000        good
 EVALUATION ERRORS              0             0        good
```

```
---- VAR hectares_v  land in production by crop-province-scen    (1000 Ha - marginal is $US per Ha)

                                LOWER          LEVEL          UPPER         MARGINAL

1-Mousil.1-wheat .normal          .           47.4000        47.4000        112.7500
1-Mousil.1-wheat .dry             .              .           47.4000          EPS
1-Mousil.2-cotton.normal          .            0.4500         0.4500        815.9200
1-Mousil.2-cotton.dry             .              .            0.4500          EPS
2-Basra .1-wheat .normal          .           16.9000        16.9000        146.2500
2-Basra .1-wheat .dry             .              .           16.9000          EPS
2-Basra .2-cotton.normal          .              .              .          1146.5000
2-Basra .2-cotton.dry             .              .              .             EPS
```

Levels:

This Model 3.1 optimizes farm income if there is a normal supply of water, equal to 800 (million cubic meters of water per year). Optimized land in production is shown here with the yellow shaded lines.

Notice the LEVELS shown for the VAR hectares_v, land in production, for each crop and province. For 3.1 you can ignore the un-shaded lines. We will see those later when model 3.2 is run for the dry water supply scenario.

What do these LEVELS say? Remember, our farmers are constrained in this model 3.1 to plant no more than the actual land planted for each crop and province. Look at the GAMS code to see what they actually planted historically. Remember this historical constraint.

Facing that constraint, our income optimizing farmers plant ___________ 47.4 (1000 ha) of wheat in Mousil, _______ 0.45 (1000 ha) of cotton in Mousil; _____________ 16.9 (1000 ha) of wheat in Basra; ______________ 0 (1000 ha) of cotton in Basra. Their optimized land in production equals their historical level for both provinces and both crops.

Marginals:

Notice the very large MARGINALS. How do you interpret them _______________?

For wheat land in Mousil, farmers are planning 47.4 (1000) ha. But if unconstrained by history, they want to plant more. If they could plant more, the 1st added (1000) ha of wheat planted in Mousil Province would earn ____________ $US 112.75 per Ha. The rest of the MARGINALS have a similar interpretation.

Notice the last line in yellow. Farmers grow zero ha of cotton in Basra. It's what they grew historically (2006). However if they could find a way to grow more cotton in Basra, their 1st added (1000) ha of cotton planted in Basra would bring them an added _______________ $US 1146.50. That is a very large amount of added farm income. So the economic value of the first extra Ha of land growing cotton in Basra is _____________ $US 1146.50 per Ha.

---- VAR T_hectares_v total land in prodn by province-scen   (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
1-Mousil.normal	.	47.8500	+INF	.
1-Mousil.dry	.	.	+INF	.
2-Basra .normal	.	16.9000	+INF	.
2-Basra .dry	.	.	+INF	.

Levels: Sums total land in production over crops for each province for the normal water supply scenario.

---- VAR uses_crop_v total water use by crop-province-scen (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
1-Mousil.1-wheat .normal	.	564.0600	+INF	.
1-Mousil.1-wheat .dry	.	.	+INF	.
1-Mousil.2-cotton.normal	.	8.1000	+INF	.
1-Mousil.2-cotton.dry	.	.	+INF	.
2-Basra .1-wheat .normal	.	228.1500	+INF	.
2-Basra .1-wheat .dry	.	.	+INF	.
2-Basra .2-cotton.normal	.	.	+INF	.
2-Basra .2-cotton.dry	.	.	+INF	.

---- VAR uses_v total water use by scen (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
normal	.	800.3100	800.3100	.
dry	.	.	400.1550	.

```
---- VAR ag_ben_k_v  total farm income by crop-province-scen     ($US 1000s - no marginals shown)

                                LOWER          LEVEL          UPPER         MARGINAL

1-Mousil.1-wheat .normal        -INF         5344.3500        +INF             .
1-Mousil.1-wheat .dry           -INF             .            +INF             .
1-Mousil.2-cotton.normal        -INF          367.1640        +INF             .
1-Mousil.2-cotton.dry           -INF             .            +INF             .
2-Basra .1-wheat .normal        -INF         2471.6250        +INF             .
2-Basra .1-wheat .dry           -INF             .            +INF             .
2-Basra .2-cotton.normal        -INF             .            +INF             .
2-Basra .2-cotton.dry           -INF             .            +INF             .
```

LEVELS: Farm Income for each crop and province: 5344 ($US 1000s) for wheat in Mousil, 367 ($US 1000) for cotton in Mousil, 2471 ($US 1000) for wheat in Basra, and nothing for cotton in Basra.

MARGINALS: Should be zero for the objective function

```
---- VAR ag_ben_v  total farm income by scen                     ($US 1000s - no marginals shown)

              LOWER          LEVEL          UPPER         MARGINAL

normal        -INF         8183.1390        +INF             .
dry           -INF             .            +INF             .
```

LEVELS: Farm Income summed over all crops and provinces = 8183 ($US 1000s)

```
                           LOWER          LEVEL          UPPER         MARGINAL

---- VAR ag_ben_no~        -INF         8183.1390        +INF             .
---- VAR ag_ben_dr~        -INF             .            +INF             .

  ag_ben_normal_v total farm income (objective - full water)       ($US 1000s - no marginals shown)
  ag_ben_dry_v    total farm income (objective - reduced water)    ($US 1000s - no marginals shown)
```

LEVELS: Farm Income for the normal water supply scenario = 8183 ($US 1000s)

3.2: REDUCED (50% OF 2006) SUPPLY

```
GAMS Rev 233  WIN-VIS 23.3.3 x86/MS Windows
05/15/13 18:46:36 Page 9
Optimizing Water Resource Use in the TE Basin - Iraq
Solution Report     SOLVE TE_03 Using NLP From line 202


               S O L V E      S U M M A R Y

     MODEL   TE_03                OBJECTIVE  ag_ben_dry_v
     TYPE    NLP                  DIRECTION  MAXIMIZE
     SOLVER  CONOPT               FROM LINE  202

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      1 Optimal
**** OBJECTIVE VALUE             4391.7544    much lower income because water available falls by half

 RESOURCE USAGE, LIMIT          0.055      1000.000
 ITERATION COUNT, LIMIT         5    2000000000
 EVALUATION ERRORS              0             0
```

---- VAR hectares_v land in production by crop-province-scen   (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
1-Mousil.1-wheat .normal	.	47.4000	47.4000	.
1-Mousil.1-wheat .dry	.	13.7735	47.4000	.
1-Mousil.2-cotton.normal	.	0.4500	0.4500	.
1-Mousil.2-cotton.dry	.	0.4500	0.4500	645.3738
2-Basra .1-wheat .normal	.	16.9000	16.9000	.
2-Basra .1-wheat .dry	.	16.9000	16.9000	18.3403
2-Basra .2-cotton.normal	.	.	.	EPS
2-Basra .2-cotton.dry	.	.	.	943.7395

```
---- VAR T_hectares_v  total land in prodn by province-scen        (1000 Ha - marginal is $US per Ha)

                       LOWER          LEVEL          UPPER         MARGINAL

1-Mousil.normal          .            47.8500        +INF              .
1-Mousil.dry             .            14.2235        +INF              .
2-Basra .normal          .            16.9000        +INF              .
2-Basra .dry             .            16.9000        +INF              .
```

```
---- VAR uses_crop_v  total water use by crop-province-scen   (million m^3 - marginal is $US per 1000 m^3)

                               LOWER          LEVEL          UPPER         MARGINAL

1-Mousil.1-wheat .normal       564.0600       564.0600       564.0600         EPS
1-Mousil.1-wheat .dry              .          163.9050         +INF            .
1-Mousil.2-cotton.normal         8.1000         8.1000         8.1000         EPS
1-Mousil.2-cotton.dry              .            8.1000         +INF            .
2-Basra .1-wheat .normal       228.1500       228.1500       228.1500         EPS
2-Basra .1-wheat .dry              .          228.1500         +INF            .
2-Basra .2-cotton.normal           .              .              .            EPS
2-Basra .2-cotton.dry              .              .            +INF            .
```

Levels

Notice LEVELS of uses_crop_v to each crop for each province. Water use falls off from 564 to 164 (million cubic meters) for wheat in Mousil in the dry scenario compared to full water. But water put to cotton stays the same for in Mousil, wheat in Basra stays the same, and cotton in Basra stays the same as for the normal water supply scenario.

Why did we reduce water applied to wheat in Mousil by so much? _____________? Hard question. It’s because wheat in Mousil earns the lowest income per ha of any crop or any province. So when water becomes scarce, wheat in Mousil drops out, but other crops and other provinces apply full water to their crops.

Marginals

MARGINALS are zero because there is no constraint on water application to any crop or to any province. But there is a big constraint on total water available. How much did total water fall with drought ______?.

```
---- VAR uses_v  total water use by scen                (million m^3 - marginal is $US per 1000 m^3)

              LOWER          LEVEL          UPPER         MARGINAL

normal          .          800.3100       800.3100          .
dry             .          400.1550       400.1550         9.4748
```

Level

Total water use over crops and scenarios falls from 800.31 (million m3) to 400.15 (million m3). A big drop off. So the MARGINAL value of water rises from 0 to $US 9.47 per 1000 cubic meter.

```
---- VAR ag_ben_k_v  total farm income by crop-province-scen     ($US 1000s - no marginals shown)

                                LOWER          LEVEL          UPPER         MARGINAL

1-Mousil.1-wheat .normal        -INF         5344.3500        +INF             .
1-Mousil.1-wheat .dry           -INF         1552.9654        +INF             .
1-Mousil.2-cotton.normal        -INF          367.1640        +INF             .
1-Mousil.2-cotton.dry           -INF          367.1640        +INF             .
2-Basra .1-wheat .normal        -INF         2471.6250        +INF             .
2-Basra .1-wheat .dry           -INF         2471.6250        +INF             .
2-Basra .2-cotton.normal        -INF              .           +INF             .


     VAR ag_ben_k_v  total farm income by crop-province-scen     ($US 1000s - no marginals shown)

                                LOWER          LEVEL          UPPER         MARGINAL

2-Basra .2-cotton.dry           -INF              .           +INF             .

---- VAR ag_ben_v  total farm income by scen                   ($US 1000s - no marginals shown)

              LOWER          LEVEL          UPPER         MARGINAL

normal        -INF         8183.1390        +INF             .
dry           -INF         4391.7544        +INF             .

                           LOWER          LEVEL          UPPER         MARGINAL

---- VAR ag_ben_no~        -INF         8183.1390        +INF             .
---- VAR ag_ben_dr~        -INF         4391.7544        +INF             .

  ag_ben_normal_v  total farm income (objective - full water)    ($US 1000s - no marginals shown)
  ag_ben_dry_v  total farm income (objective - reduced water) ($US 1000s - no marginals shown)
```

Model 4

Code

```
$EOLCOM //
$TITLE Afghan Water Rights Priority Model
$OFFSYMXREF OFFSYMLIST OffLISTING OFFUPPER

OPTION LIMROW=000, LIMCOL = 0;

$ONTEXT

EOLCOM // tells GAMS to ignore anything in the line's text after the symbol //
OFFLISTING deletes all program lines and just includes GAMS listing
Set LIMROW = 0 to eliminate all equations in the GAMS listing
Set LIMROW = 100 or more to show all equations in listing. Helps trap  errors

* ------------------------------------------------------------------------------------------
  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  Dina A. Salman:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, New Mexico USA
  email: dinasalm@nmsu.edu

  Saud A.Amer:
  US Geological Survey - International Water Resources Division
  Reston VA  USA
  email:  samer@usgs.gov

  May 13, 2013

* ------------------------------------------------------------------------------------------
*  Model 4 Simple Farm management
*  Adds a SET (r) for water-sharing rule based on priority allocation of water as parameter
*  water sharing rule requires introducing a priority SET (j)
*  Also solves several models automatically inside a series of loops
* ------------------------------------------------------------------------------------------

$OFFTEXT

****************************************   Section 1 ***********************************
*                                           Sets                                          *
```

```
*************************************************************************************

SETS

i    province

/
  1-Mousil
  2-Basra
/

k    crop

/  1-wheat
   2-cotton
/

s   hydrologic water supply scenario

/
normal
dry
/

j  water right priority: same number # of elements as # of provinces - 1 priority per province

/       j1*j2
/

r water allocation rule label: allows evaluating many water shortage sharing rules
/
us_priority     upstream   priority                  //rule 1
prop_sharing    proportional sharing of shortages    //rule 2
/

set rji(r,j,i) mapping set: assigns water use priorities to province - separate mapping for each proposed rule

/ us_priority  . (j1.1-Mousil, j2.2-Basra)
  prop_sharing . j1. (1-Mousil,   2-Basra)
/

display rji;
```

```
*************************************  Section 2
****************************************************************************
*                                         Data
*
*************************************************************************************************************************
****

table  Bc(i,k)    crop water demand(ET) (1000s cubic meters per Ha = 10ths of meters depth)

                 1-wheat          2-cotton
 1-mousil          11.9             18.0
 2-Basra           13.5             21.4

*Bc(k) =  0.5 * Bc(k)

table    Yield_p(i,k)   Crop Yield tons per Ha (proportional to ET)

               1-wheat         2-Cotton
1-Mousil        1.39             2.32
2-Basra         1.57             2.75


* economic data


parameter   Price_p(k)        Crop Prices ($ US per ton)

/1-Wheat        225
 2-Cotton       906
/

table Cost_p(i, k)     Crop Production Costs excluding water ($ US per Ha)

                1-wheat   2-cotton
1-Mousil          200      1286
2-Basra           207      1345

;

* data from Dr. Saud Amer 2012
```

```
Table land_p(i,k)   observed land in prodn year 2006 (1000 Ha)

                 1-wheat        2-cotton
1-mousil           47.4           0.45
2-basra            16.9           0.00
;

Parameter Net_revenue_p(i,k)  calculated net revenue per unit land ($ US per HA)
;

Net_revenue_p(i,k) = Price_p(k) * Yield_p(i,k) - Cost_p(i,k);

parameter Wat_supply_p(s)      total water available (million cubic meters per year)
;

Wat_supply_p('normal') = sum((i,k), land_p(i,k) * Bc(i,k));   // calculates observed water use
Wat_supply_p('dry')    = 0.5 * wat_supply_p('normal');        // dry: half of observed water use

parameter right_p(i)  total water assigned by province in full supply conditions (paper water)
;

right_p(i)  =  sum(k, Bc(i,k) * land_p(i,k));  // sums over crops: ith province's historical (2006) water use under
full supply

* code below loops over each priority (j) then maps each priority to corresponding province i
* code is thanks to Pete Stacy at GAMS Development Corporation Feb 1 2012

parameter   tot_assigned(r,s) cumulative water assigned to province (i) including higher priorities;
            tot_assigned(r,s)  = 0;  // starts at 0 - prepares to loop

parameter
  tot_pap_right_by_prior(r,  j) cumulative paper right at jth priority for all provinces including ties
  remain_supply_by_prior(r,s,j) residual supply    at jth priority after supplying higher priorities
  wet_wat_use           (r,s,i) wet (not paper) water use assigned to ith province - limited by total basin supply
;

Loop(r,                                                                      // loop over sharing rule (r)
 Loop(j,                                                                     // loop over priority (j)

   tot_pap_right_by_prior(r,j) = sum[i$rji(r, j, i), right_p(i)] ;          // total paper rights by (j) priority by
provinces includes ties
   remain_supply_by_prior(r,s,j) =
```

```
      min[(wat_supply_p(s) - tot_assigned(r,s)), tot_pap_right_by_prior(r,j)]; // remaining supply by (j) priority
after protecting higher priorities
   Loop(i$rji(r,j,i),
       wet_wat_use(r,s,i) = (right_p(i)/tot_pap_right_by_prior(r,j))
       * remain_supply_by_prior(r,s,j) + eps;                           // wet water assigned to (i) province
    tot_assigned(r,s) = tot_assigned(r,s) + wet_wat_use(r,s,i) ;        // cumulative water (check) assigned to
last province getting water - should match total supply

   );                                                                   // end province loop
  );                                                                    // end priority loop
);                                                                      // end rule loop

parameter tot_wat_use(r,s)  total water use defined;
tot_wat_use  (r,s) = sum(i, wet_wat_use(r,s,i)) + eps;  // checks total use assigned against supply

*********** end of water allocation system ******************

**** begin looping over water sharing rule and water supply scenario below
* rr, ss, are subset of original sets... allows fast scenario analysis below

set rr(r);   // water sharing rule
set ss(s);   // water supply scenario

rr(r)=no;    // switches subset off for now
ss(s)=no;    // switches subset off for now

******************************  SECTION 3 ***********************************************
*                               Variables                                               *
*****************************************************************************************

positive variables

hectares_v    (r,i,k,s)   land in production by rule-province-crop-scen    (1000 Ha - marginal is $US per Ha)
T_hectares_v  (r,i,  s)   total land in prodn by rule-province-scen        (1000 Ha - marginal is $US per Ha)
uses_crop_v   (r,i,k,s)   total water use by rule-province-crop-scen       (million m^3 - marginal is $US per 1000 m^3)
uses_v        (r,i,  s)   total water use by rule-province-scen            (million m^3 - marginal is $US per 1000 m^3)

variables

ag_ben_k_v    (r,i,k,s)   total farm income by rule-province-crop-scen     ($US 1000s - no marginals shown)
ag_ben_v      (r,    s)   total farm income by rule-scen                   ($US 1000s - no marginals shown)
```

```
Tot_b_v          total farm income re-calc for each rule-scen (objective) ($US 1000s - no marginals shown)

*******************************   Section 4  ********************************************
*                                 Equations                                               *
*****************************************************************************************

Equations

// Equations DECLARED

T_hectares_e   (r,i,  s)    total land in production by province-scen  (1000 Ha - marginal is $US per Ha)
Uses_crop_e    (r,i,k,s)    total water use by crop-province-scean     (million m^3 - marginal is $US per 1000 m^3)
uses_e         (r,i,  s)    total water use by scen                    (million m^3 - marginal is $US per 1000 m^3)

ag_ben_k_e     (r,i,k,s)    total farm income by crop-province-scen    ($US 1000s - no marginals shown)
ag_ben_e       (r,    s)    total farm income by scen                  ($US 1000s - no marginals shown)

Tot_b_e                     total farm income by element of loop       ($US 1000s - no marginals shown)

;

// Equations defined (using above names with algebraic formulas)

// Equations below defined over rule (rr) and scenario (ss) using subset rr and ss

T_hectares_e   (rr,i,  ss)..  T_hectares_v(rr,i,  ss)   =e=  sum(k,   hectares_v  (rr,i,k,ss));
Uses_crop_e    (rr,i,k,ss)..  uses_crop_v (rr,i,k,ss)   =e=  Bc(i,k) *  hectares_v(rr,i,k,ss);
uses_e         (rr,i,  ss)..  uses_v      (rr,i,  ss)   =e=  sum(k,    uses_crop_v(rr,i,k,ss));

ag_ben_k_e     (rr,i,k,ss)..  ag_ben_k_v  (rr,i,k,ss)   =e=  net_revenue_p(i,k) * hectares_v(rr,i,k,ss);
ag_ben_e       (rr,    ss)..  ag_ben_v    (rr,    ss)   =e=  sum((i,k), ag_ben_k_v(rr,i,k,ss));

Tot_b_e                   ..  Tot_b_v                   =e=  sum((rr,ss), ag_ben_v(rr,ss));   // sums total benefits
over indices for sensitivity analysis

**************** Section 5 **************************************************************
*  Labels and defines models                                                              *
*  Each model has one objective function                                                  *
*****************************************************************************************

model TE_04 /all/ ;
```

```
*********************************  Section 6   ****************************************
*                                  BOUNDS                                              *
*  Bounding shows a positive shadow price for each limiting resource                   *
***************************************************************************************

uses_v.up    (r,i,  s) = wet_wat_use(r,s,i  );  // upper bound limits available water to observed water use
hectares_v.up(r,i,k,s) =      land_p(    i,k);  // can produce no more land by crop than under full water supply

*hectares_v.lo('1-mousil','1-wheat') =  10;     // food security requirement
*uses_crop_v.lo('wheat')  = 1;                  // another way of enforcing food security

**************** Section 7 *************************************************************
*               SOLVE THE MODEL                                                           *
***************************************************************************************

  loop(r,
   loop(s,

     ss(s) = yes;
     rr(r) = yes;

     Solve TE_04 using nlp maximizing Tot_b_v;  // multiple solves inside loop avoids writing solve several times

* closes loops below over r,s (rule for sharing shortages, water supply scenario)

    rr(r) = no;
    ss(s) = no;

   );
  );

// THE END
```

Model 4

Listing

MODEL 4: ADDS 2 WATER SHORTAGE SHARING METHODS TO MODEL 3

- **OPTIMIZES FARM INCOME FOR FULL AND REDUCED WATER SUPPLY**
- **FOR UPSTREAM PRIORITY AND PROPORTIONAL SHARING OF SHORTAGES**

4 MODELS TOTAL

4.1: FULL (100% OF 2006) SUPPLY
UPSTREAM PRIORITY FOR SHARING SHORTAGES

```
          L O O P S                  r us_priority
                                     s normal

          S O L V E     S U M M A R Y

     MODEL   TE_04               OBJECTIVE  Tot_b_v
     TYPE    NLP                 DIRECTION  MAXIMIZE
     SOLVER  CONOPT              FROM LINE  266

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      1 Optimal
**** OBJECTIVE VALUE             8183.1390

 RESOURCE USAGE, LIMIT          0.016      1000.000
 ITERATION COUNT, LIMIT         5    2000000000
 EVALUATION ERRORS              0             0
```

---- VAR hectares_v land in production by rule-province-crop-scen (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.1-wheat .normal	.	47.4000	47.4000	112.7500
us_priority.1-Mousil.2-cotton.normal	.	0.4500	0.4500	815.9200
us_priority.2-Basra .1-wheat .normal	.	16.9000	16.9000	146.2500
us_priority.2-Basra .2-cotton.normal	.	.	.	1146.5000

---- VAR T_hectares_v total land in prodn by rule-province-scen (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.normal	.	47.8500	+INF	.
us_priority.2-Basra .normal	.	16.9000	+INF	.

---- VAR uses_crop_v total water use by rule-province-crop-scen (million m^3 - marginal $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.1-wheat .normal	.	564.0600	+INF	.
us_priority.1-Mousil.2-cotton.normal	.	8.1000	+INF	.
us_priority.2-Basra .1-wheat .normal	.	228.1500	+INF	.
us_priority.2-Basra .2-cotton.normal	.	.	+INF	.

---- VAR uses_v total water use by rule-province-scen   (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.normal	.	572.1600	572.1600	.
us_priority.2-Basra .normal	.	228.1500	228.1500	.

---- VAR ag_ben_k_v total farm income by rule-province-crop-scen   ($US 1000s - no marginals shown)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.1-wheat .normal	-INF	5344.3500	+INF	.
us_priority.1-Mousil.2-cotton.normal	-INF	367.1640	+INF	.
us_priority.2-Basra .1-wheat .normal	-INF	2471.6250	+INF	.
us_priority.2-Basra .2-cotton.normal	-INF	.	+INF	.

```
---- VAR ag_ben_v  total farm income by rule-scen           ($US 1000s - no marginals shown)

                          LOWER          LEVEL          UPPER         MARGINAL

us_priority.normal        -INF         8183.1390        +INF              .

                           LOWER          LEVEL          UPPER         MARGINAL

---- VAR Tot_b_v           -INF         8183.1390        +INF              .

  Tot_b_v  total farm income re-calc for each rule-scen (objective) ($US 1000s - no marginals shown)
```

4.2: REDUCED (50% OF 2006) SUPPLY
UPSTREAM PRIORITY FOR SHARING SHORTAGES

```
             L O O P S                  r us_priority
                                        s dry

             S O L V E      S U M M A R Y

     MODEL   TE_04                OBJECTIVE  Tot_b_v
     TYPE    NLP                  DIRECTION  MAXIMIZE
     SOLVER  CONOPT               FROM LINE  266

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      1 Optimal
**** OBJECTIVE VALUE             4081.8028

 RESOURCE USAGE, LIMIT          0.000      1000.000
 ITERATION COUNT, LIMIT         5    2000000000
 EVALUATION ERRORS              0             0
```

---- VAR hectares_v land in production by rule-province-crop-scen   (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.1-wheat .dry	.	32.9458	47.4000	.
us_priority.1-Mousil.2-cotton.dry	.	0.4500	0.4500	645.3738
us_priority.2-Basra .1-wheat .dry	.	.	16.9000	.
us_priority.2-Basra .2-cotton.dry	.	.	.	914.6667

---- VAR T_hectares_v total land in prodn by rule-province-scen   (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.dry	.	33.3958	+INF	.
us_priority.2-Basra .dry	.	.	+INF	.

---- VAR uses_crop_v total water use by rule-province-crop-scen   (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.1-wheat .dry	.	392.0550	+INF	.
us_priority.1-Mousil.2-cotton.dry	.	8.1000	+INF	.
us_priority.2-Basra .1-wheat .dry	.	.	+INF	.
us_priority.2-Basra .2-cotton.dry	.	.	+INF	.

---- VAR uses_v total water use by rule-province-scen   (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.dry	.	400.1550	400.1550	9.4748
us_priority.2-Basra .dry	.	EPS	EPS	10.8333

---- VAR ag_ben_k_v total farm income by rule-province-crop-scen   ($US 1000s - no marginals shown)

	LOWER	LEVEL	UPPER	MARGINAL
us_priority.1-Mousil.1-wheat .dry	-INF	3714.6388	+INF	.
us_priority.1-Mousil.2-cotton.dry	-INF	367.1640	+INF	.
us_priority.2-Basra .1-wheat .dry	-INF	.	+INF	.
us_priority.2-Basra .2-cotton.dry	-INF	.	+INF	.

```
---- VAR ag_ben_v  total farm income by rule-scen          ($US 1000s - no marginals shown)

                       LOWER          LEVEL          UPPER         MARGINAL

us_priority.dry         -INF          4081.8028        +INF             .

                           LOWER          LEVEL          UPPER         MARGINAL

---- VAR Tot_b_v           -INF          4081.8028        +INF             .

  Tot_b_v  total farm income re-calc for each rule-scen (objective) ($US 1000s - no marginals shown)
```

4.3: RETURN TO FULL (100% OF 2006) SUPPLY
PROPORTIONAL SHARING OF SHORTAGES

```
           L O O P S                  r prop_sharing
                                      s normal

           S O L V E      S U M M A R Y

     MODEL   TE_04               OBJECTIVE  Tot_b_v
     TYPE    NLP                 DIRECTION  MAXIMIZE
     SOLVER  CONOPT              FROM LINE  266

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      1 Optimal
**** OBJECTIVE VALUE             8183.1390

 RESOURCE USAGE, LIMIT          0.016      1000.000
 ITERATION COUNT, LIMIT         5    2000000000
 EVALUATION ERRORS              0             0
```

```
---- VAR hectares_v  land in production by rule-province-crop-scen    (1000 Ha - marginal is $US per Ha)

                                             LOWER          LEVEL          UPPER         MARGINAL

prop_sharing.1-Mousil.1-wheat .normal          .            47.4000        47.4000        112.7500
prop_sharing.1-Mousil.2-cotton.normal          .             0.4500         0.4500        815.9200
prop_sharing.2-Basra .1-wheat .normal          .            16.9000        16.9000        146.2500
prop_sharing.2-Basra .2-cotton.normal          .              .              .           1146.5000

---- VAR T_hectares_v  total land in prodn by rule-province-scen     (1000 Ha - marginal is $US per Ha)

                                    LOWER          LEVEL          UPPER         MARGINAL

prop_sharing.1-Mousil.normal          .            47.8500        +INF             .
prop_sharing.2-Basra .normal          .            16.9000        +INF             .

---- VAR uses_crop_v  total water use by rule-province-crop-scen   (million m^3 - marginal is $US per 1000 m^3)

                                             LOWER          LEVEL          UPPER         MARGINAL

prop_sharing.1-Mousil.1-wheat .normal          .           564.0600        +INF             .
prop_sharing.1-Mousil.2-cotton.normal          .             8.1000        +INF             .
prop_sharing.2-Basra .1-wheat .normal          .           228.1500        +INF             .
prop_sharing.2-Basra .2-cotton.normal          .              .            +INF             .

---- VAR uses_v  total water use by rule-province-scen     (million m^3 - marginal is $US per 1000 m^3)

                                    LOWER          LEVEL          UPPER         MARGINAL

prop_sharing.1-Mousil.normal          .           572.1600       572.1600          .
prop_sharing.2-Basra .normal          .           228.1500       228.1500          .

---- VAR ag_ben_k_v  total farm income by rule-province-crop-scen     ($US 1000s - no marginals shown)

                                             LOWER          LEVEL          UPPER         MARGINAL

prop_sharing.1-Mousil.1-wheat .normal        -INF         5344.3500        +INF             .
prop_sharing.1-Mousil.2-cotton.normal        -INF          367.1640        +INF             .
prop_sharing.2-Basra .1-wheat .normal        -INF         2471.6250        +INF             .
prop_sharing.2-Basra .2-cotton.normal        -INF             .            +INF             .
```

```
---- VAR ag_ben_v  total farm income by rule-scen           ($US 1000s - no marginals shown)

                           LOWER          LEVEL          UPPER         MARGINAL

prop_sharing.normal        -INF         8183.1390        +INF              .

                           LOWER          LEVEL          UPPER         MARGINAL

---- VAR Tot_b_v           -INF         8183.1390        +INF              .

  Tot_b_v  total farm income re-calc for each rule-scen (objective) ($US 1000s - no marginals shown)
```

4.4: REDUCED (50% OF 2006) SUPPLY AND PROPORTIONAL SHARING OF SHORTAGES

```
            L O O P S                 r prop_sharing
                                      s dry

            S O L V E     S U M M A R Y

     MODEL   TE_04               OBJECTIVE  Tot_b_v
     TYPE    NLP                 DIRECTION  MAXIMIZE
     SOLVER  CONOPT              FROM LINE  266

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      1 Optimal
**** OBJECTIVE VALUE             4236.7786

 RESOURCE USAGE, LIMIT          0.016      1000.000
 ITERATION COUNT, LIMIT         5    2000000000
 EVALUATION ERRORS              0             0
```

---- VAR hectares_v land in production by rule-province-crop-scen   (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
prop_sharing.1-Mousil.1-wheat .dry	.	23.3597	47.4000	.
prop_sharing.1-Mousil.2-cotton.dry	.	0.4500	0.4500	645.3738
prop_sharing.2-Basra .1-wheat .dry	.	8.4500	16.9000	.
prop_sharing.2-Basra .2-cotton.dry	.	.	.	914.6667

---- VAR T_hectares_v total land in prodn by rule-province-scen   (1000 Ha - marginal is $US per Ha)

	LOWER	LEVEL	UPPER	MARGINAL
prop_sharing.1-Mousil.dry	.	23.8097	+INF	.
prop_sharing.2-Basra .dry	.	8.4500	+INF	.

---- VAR uses_crop_v total water use by rule-province-crop-scen   (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
prop_sharing.1-Mousil.1-wheat .dry	.	277.9800	+INF	.
prop_sharing.1-Mousil.2-cotton.dry	.	8.1000	+INF	.
prop_sharing.2-Basra .1-wheat .dry	.	114.0750	+INF	.
prop_sharing.2-Basra .2-cotton.dry	.	.	+INF	.

---- VAR uses_v total water use by rule-province-scen   (million m^3 - marginal is $US per 1000 m^3)

	LOWER	LEVEL	UPPER	MARGINAL
prop_sharing.1-Mousil.dry	.	286.0800	286.0800	9.4748
prop_sharing.2-Basra .dry	.	114.0750	114.0750	10.8333

---- VAR ag_ben_k_v total farm income by rule-province-crop-scen   ($US 1000s - no marginals shown)

	LOWER	LEVEL	UPPER	MARGINAL
prop_sharing.1-Mousil.1-wheat .dry	-INF	2633.8021	+INF	.
prop_sharing.1-Mousil.2-cotton.dry	-INF	367.1640	+INF	.
prop_sharing.2-Basra .1-wheat .dry	-INF	1235.8125	+INF	.
prop_sharing.2-Basra .2-cotton.dry	-INF	.	+INF	.

```
---- VAR ag_ben_v  total farm income by rule-scen         ($US 1000s - no marginals shown)

                        LOWER          LEVEL          UPPER         MARGINAL

prop_sharing.dry        -INF         4236.7786        +INF              .

                           LOWER          LEVEL          UPPER         MARGINAL

---- VAR Tot_b_v           -INF         4236.7786        +INF              .

  Tot_b_v  total farm income re-calc for each rule-scen (objective) ($US 1000s - no marginals shown)
```

Model 5

Code

```
$EOLCOM //
$TITLE Iraq Water Rights Priority Model
$OFFSYMXREF OFFSYMLIST OffLISTING OFFUPPER

OPTION LIMROW=000, LIMCOL = 0;

$ONTEXT

EOLCOM // tells GAMS to ignore anything in the line's text after the symbol //
OFFLISTING deletes all program lines and just includes GAMS listing
Set LIMROW = 0 to eliminate all equations in the GAMS listing
Set LIMROW = 100 or more to show all equations in listing. Helps trap  errors

* ---------------------------------------------------------------------------------------
  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  Dina A. Salman:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, New Mexico USA
  email: dinasalm@nmsu.edu

  Saud A.Amer:
  US Geological Survey - International Water Resources Division
  Reston VA  USA
  email:  samer@usgs.gov

  May 13, 2013

* ---------------------------------------------------------------------------------------
*  Model 5 Simple Farm management
*  Exports results to excel spreadsheet
* ---------------------------------------------------------------------------------------

$OFFTEXT

****************************************   Section 1 ***********************************
*                                           Sets                                       *
****************************************************************************************
```

```
SETS

i    province

/
  1-Mousil
  2-Basra
/

k    crop

/  1-wheat
   2-cotton
/


s   hydrologic water supply scenario

/
normal
dry/

j  water right priority:  water right priority: same number # of elements as # of provinces - 1 priority per province

/      j1*j2
/

r water allocation rule allows for many water shortage sharing rules to be considered

/
us_priority         upstream   priority                   //rule 1
prop_sharing        proportional sharing of shortages     //rule 2
/

set rji(r, j, i) mapping set: assigns priorities to province - separate assignments for each proposed rule

/ us_priority  .  (j1.1-Mousil,  j2.2-Basra)
  prop_sharing .  (j1.1-Mousil,  j1.2-Basra)
/

************************************  Section 2
**************************************************************************
```

```
*                                          Data
*
************************************************************************************************************************
****

table  Bc(i,k)    crop water demand (ET) (1000s cubic meters per Ha = 10ths of meters depth)

* Data: Allen 1998, FAO 56 applied to year 2012 weather information in Iraq
* Crop Water Requirements: Food and Agricultural Organization Report 56.

                 1-wheat          2-cotton
 1-mousil          11.9             18.0
 2-Basra           13.5             21.4

*Bc(k) =  0.5 * Bc(k)

table    Yield_p(i,k)   Crop Yield tons per Ha (proportional to ET)

               1-wheat         2-Cotton
1-Mousil        1.39             2.32
2-Basra         1.57             2.75

* economic data


parameter   Price_p(k)        Crop Prices ($ US per ton)

/1-Wheat        225
 2-Cotton       906
/

table Cost_p(i, k)    Crop Production Costs excluding water ($ US per Ha)

* FAO Ag Stats, adapted from Iran and Egypt

                1-wheat   2-cotton
1-Mousil          200      1286
2-Basra           207      1345

;

* data from Dr. Saud Amer 2012
```

```
Table land_p(i,k)   observed land in prodn for year 2006 (1000 Ha)

                 1-wheat        2-cotton
1-mousil           47.4           0.45
2-basra            16.9           0.00
;

Parameter Net_revenue_p(i,k)  calculated net revenue per ha ($ US per HA)
;

Net_revenue_p(i,k) = Price_p(k) * Yield_p(i,k) - Cost_p(i,k);

parameter Wat_supply_p(s)      total water available (million cubic meters per year) calculated total ag water use (not
meas from gauges)
;
Wat_supply_p('normal') = sum((i,k), land_p(i,k) * Bc(i,k));  // calculates observed water use
Wat_supply_p('dry'   ) = 0.50 * wat_supply_p('normal');      // dry: half of observed value

parameter right_p(i)    total water assigned by province in full supply conditions (paper water)
;

right_p(i)  =  sum(k, Bc(i,k) * land_p(i,k));  // sums over crops: ith province's historical (2006) water use under
full supply

*******************  code defines water sharing methods *****************************************

* code below loops over each priority (j) then maps each priority to corresponding province i
* code is thanks to Pete Stacy at GAMS Development Corporation Feb 1 2012

parameter   tot_assigned(r,s)   cumulative water assigned to province (i) including higher priorities;
            tot_assigned(r,s)  = 0;  // starts at 0 - prepares to loop

parameter
  tot_pap_right_by_prior(  r,j)  cumulative paper right at jth priority for all provinces including ties
  remain_supply_by_prior(r,s,j)  residual supply    at jth priority after supplying higher priorities
  wet_wat_use           (r,s,i)  wet (not paper) water use assigned to ith province - limited by total basin supply
;

Loop(r,                                                                          // loop over sharing rule (r)
 Loop(j,                                                                         // loop over priority (j)
```

```
   tot_pap_right_by_prior(r,j) = sum[i$rji(r, j, i), right_p(i)] ;           // total paper rights by (j) priority by
provinces includes ties
   remain_supply_by_prior(r,s,j) =
       min[(wat_supply_p(s) - tot_assigned(r,s)), tot_pap_right_by_prior(r,j)] + eps;    // remaining supply by (j)
priority after protecting higher priorities
   Loop(i$rji(r, j, i),
       wet_wat_use(r,s,i) = (right_p(i)/tot_pap_right_by_prior(r,j))
       * remain_supply_by_prior(r,s,j) + eps;                                 // wet water assigned to (i) province
    tot_assigned(r,s) = tot_assigned(r,s) + wet_wat_use(r,s,i) ;              // cumulative water (check) assigned to
last province getting water - should match total supply

   );                                                                         // end province loop
  );                                                                          // end priority loop
);                                                                            // end rule loop

parameter tot_wat_use(r,s)  total water use defined;
tot_wat_use  (r,s) = sum(i, wet_wat_use(r,s,i)) + eps;   // checks total use assigned against supply

*********** end of water allocation system ************************************************************

**** begin looping over water sharing rule and water supply scenario below
* rr, ss, are subset of orginal sets... allows fast scenario analysis below


set rr(r);   // water sharing rule
set ss(s);   // water supply scenario

rr(r)=no;    // switches subset off for now
ss(s)=no;    // switches subset off for now

******************************  SECTION 3 *********************************************
*                                Variables                                            *
*************************************************************************************

positive variables

hectares_v        (r,s,i,k)      land in production by rule-province-crop-scen    (1000 Ha - marginal is $US per Ha)
T_hectares_v      (r,s,i  )      total land in prodn by rule-province-scen        (1000 Ha - marginal is $US per Ha)

uses_crop_v       (r,s,i,k)      total water use by rule-province-crop-scen       (million m^3 - marginal is $US per
1000 m^3)
uses_v            (r,s,i  )      total water use by rule-province-scen            (million m^3 - marginal is $US per
```

```
1000 m^3)

variables

ag_ben_k_v        (r,s,i,k)      total farm income by rule-province-crop-scen     ($US 1000s - no marginals shown)
ag_ben_v          (r,    s)      total farm income by rule-scen                   ($US 1000s - no marginals shown)

Tot_b_v          total farm income re-calc for each rule-scen (objective) ($US 1000s - no marginals shown)

*******************************   Section 4  ********************************************
*                                 Equations                                              *
******************************************************************************************


Equations

// Equations DECLARED

T_hectares_e       (r,s,i  )    total land in production by province-scen  (1000 Ha - marginal is $US per Ha)

Uses_crop_e        (r,s,i,k)    total water use by crop-province-scean     (million m^3 - marginal is $US per 1000 m^3)
uses_e             (r,s,i  )    total water use by scen                    (million m^3 - marginal is $US per 1000 m^3)

ag_ben_k_e         (r,s,i,k)    total farm income by crop-province-scen    ($US 1000s - no marginals shown)
ag_ben_e           (r,    s)    total farm income by scen                  ($US 1000s - no marginals shown)

Tot_b_e                         total farm income by element of loop       ($US 1000s - no marginals shown)
;


// Equations defined (using above names with algebraic formulas)

// Equations below defined over rule (rr) and scenario (ss) using subset rr and ss

T_hectares_e (rr,ss,i  )..  T_hectares_v(rr,ss,i  )  =e=  sum(k,   hectares_v  (rr,ss,i,k));

Uses_crop_e  (rr,ss,i,k)..  uses_crop_v (rr,ss,i,k)  =e=  Bc(i,k) *  hectares_v(rr,ss,i,k );
uses_e       (rr,ss,i  )..  uses_v      (rr,ss,i  )  =e=  sum(k,    uses_crop_v(rr,ss,i,k));

ag_ben_k_e   (rr,ss,i,k)..  ag_ben_k_v  (rr,ss,i,k)  =e=  net_revenue_p(i,k) * hectares_v(rr,ss,i,k);
ag_ben_e     (rr,ss    )..  ag_ben_v    (rr,ss)      =e=  sum((i,k), ag_ben_k_v(rr,ss,i,k));
```

```
Tot_b_e                  ..  Tot_b_v                  =e=  sum((rr,ss), ag_ben_v(rr,ss   ));   // sums total benefits
over indices for sensitivity analysis

**************** Section 5 *****************************************************************
*  Labels and defines models                                                                *
*  Each model has one objective function                                                    *
*********************************************************************************************

model TE_05 /all/ ;

************************************  Section 6   *****************************************
*                                    BOUNDS                                                 *
*  Bounding shows a positive shadow price for each limiting resource
*********************************************************************************************

uses_v.up    (r,s,i  ) = wet_wat_use(r,s,i  );  // upper bound limits available water to observed water use
hectares_v.up(r,s,i,k) =      land_p(    i,k);  // can produce no more crops than observed under full water supply
conditions

*hectares_v.lo('1-mousil','1-wheat') =  10;     // food security requirement
*uses_crop_v.lo('wheat')  = 1;                  // another way of enforcing food security


**************** Section 7 *****************************************************************
*               SOLVE THE MODEL                                                             *
*********************************************************************************************
parameter shad_price_p(r,s,i)   names shadow price of water to send to spreadsheet
;

  loop(r,
   loop(s,

     ss(s) = yes;
     rr(r) = yes;

     Solve TE_05 using nlp maximizing Tot_b_v;

      shad_price_p(r,s,i) =       uses_v.m(r,s,i) + eps;    // export shad prices for all water allocation rules except
unrestricted trading

* closes loops below over r,s (rule for sharing shortages, water supply scenario)
```

```
    rr(r) = no;
    ss(s) = no;

   );
  );


*************************************************************************************
*   Section 8: DISPLAY RESULTS TO SPREADSHEET                                      *
*************************************************************************************

parameter

uses_p                      (r,s,i  )   total water use by province summed over crops
land_v_p                    (r,s,i,k)   total land in production
ben_by_crop_province_p      (r,s,i,k)   total benefits by crop and province
ben_by_province_p           (r,s,i  )   total economic benefits by province
tot_ben_p                   (r,s    )   total economic benefits

;

* land in prodn (1000 hectares)

land_v_p                    (r,s,i,k) = hectares_v.l(r,s,i,k) + eps;

* water use (million cubic meters)

uses_p                      (r,s,i  ) =    Uses_v.l(r,s,i) + eps;

* economic benefits ($1000 US)

ben_by_crop_province_p      (r,s,i,k)     =        ag_ben_k_v.l(r,s,i,k ) + eps;
ben_by_province_p           (r,s,i  )     = sum(k, ag_ben_k_v.l(r,s,i,k)) + eps;
tot_ben_p                   (r,s    )     =         ag_ben_v.l (r,s     ) + eps;

* GAMS GDX facility writes to excel spreadsheet

execute_unload "farm_mgmt_05.gdx"

tot_pap_right_by_prior   remain_supply_by_prior wet_wat_use
tot_assigned             land_p                 land_v_p
uses_p                   ben_by_crop_province_p ben_by_province_p
```

```
tot_ben_p                net_revenue_p          shad_price_p
Price_p                  yield_p                cost_p
Net_revenue_p            Bc

;

$onecho > gdxxrwout.txt

i=farm_mgmt_05.gdx
o=farm_mgmt_05.xls

epsout = 0

par = tot_pap_right_by_prior  rng =  total_paper_right_by_priority!c4   Cdim = 0
par = remain_supply_by_prior  rng =  remain_supply_by_priority!c4       cdim = 0
par = wet_wat_use             rng =  wet_water_use_by_province!c4       Cdim = 0
par = tot_assigned            rng =  basin_cum_water_use!c4             cdim = 0

par = land_v_p                rng =  land_in_prodn!c4                   cdim = 0
par = uses_p                  rng =  total_water_use_by_province!c4     cdim = 0

par = net_revenue_p           rng =  net_rev_per_ha!c4                  cdim = 0
par = ben_by_crop_province_p  rng =  benefits_by_crop_province!c4       cdim = 0
par = ben_by_province_p       rng =  benefits_by_province!c4            cdim = 0
par = tot_ben_p               rng =  total_benefits!c4                  cdim = 0

par = shad_price_p            rng =  shadow_price!c4                    cdim = 0
par = Price_p                 rng =  price!c4                           cdim = 0
par = cost_p                  rng =  cost!c4                            cdim = 0
par = yield_p                 rng =  yield!c4                           cdim = 0
par = Net_revenue_p           rng =  net_rev!c4                         cdim = 0
par = Bc                      rng =  wu_ha!c4                           cdim = 0

$offecho
execute 'gdxxrw.exe @gdxxrwout.txt trace=2';

// THE END
```

Model 5

Tabled Results

Table 1: Average Price by Crop, Lower Tigris-Euphrates, Iraq, 2006 ($US / ton)	
Crop	Price
1-wheat	225
2-cotton	906

Table 2: Baseline Crop Yield by Province and Crop, Lower Tigris-Euphrates, Iraq, 2006 (metric tons/ha)		
Province	Crop	Yield
1-Mousil	1-wheat	1.39
	2-cotton	2.32
2-Basra	1-wheat	1.57
	2-cotton	2.75

Table 3: Production Cost per Ha, by Province and Crop, Lower Tigris-Euphrates, Iraq, 2006 ($US per ha)		
Province	Crop	Total
1-Mousil	1-wheat	200
	2-cotton	1286
2-Basra	1-wheat	207
	2-cotton	1345

Table 4: Average Net Farm Income per Ha by Crop and Province, Lower Tigris-Euphrates, Iraq, 2006 base conditions ($US per ha)		
Province	Crop	Net Revenue Per Ha
1-Mousil	1-wheat	112.75
	2-cotton	815.92
2-Basra	1-wheat	146.25
	2-cotton	1146.5

Table 05: Calculated crop water consumption in 10ths of meters depth per Ha, Iraq weather data of 2012.		
Province	Crop	Depth
1-Mousil	1-wheat	11.9
	2-cotton	18
2-Basra	1-wheat	13.5
	2-cotton	21.4

Table 6: Regional Farm Income by Water Sharing Arrangement, Water Supply, and Province, Lower Tigris-Euphrates Basin, Iraq, 2006 ($US 1000/year)			
Province	Water Sharing Rule	Water Supply condition	Income
1-Mousil	us_priority	normal	5,712
		dry	4,082
	prop_sharing	normal	5,712
		dry	3,001
2-Basra	us_priority	normal	2,472
		dry	0
	prop_sharing	normal	2,472
		dry	1,236

Table 7: Irrigated Land in Production, by Province, Crop, Shortage Sharing Arrangement, and Water Supply Scenario, Tigris and Euphrates Basins, Iraq, 2006, (1000 hectares/year)				
Crop	Province	Water Sharing Rule	Water Supply Condition	land in production
1-wheat	1-Mousil	us_priority	normal	47.40
			dry	32.95
		prop_sharing	normal	47.40
			dry	23.36
	2-Basra	us_priority	normal	16.90
			dry	0.00
		prop_sharing	normal	16.90
			dry	8.45
2-cotton	1-Mousil	us_priority	normal	0.45
			dry	0.45
		prop_sharing	normal	0.45
			dry	0.45
	2-Basra	us_priority	normal	0.00
			dry	0.00
		prop_sharing	normal	0.00
			dry	0.00

Table 8: Total Water Use Summed over Crops, by Province, Crop, Shortage Sharing Arrangement, and Water Supply Scenario, Tigris and Euphrates Basins, Iraq, 2006, (million cubic meters/year)

Province	Water Sharing Rule	Water Supply Condition	Total
1-Mousil	us_priority	normal	572.16
		dry	400.155
	prop_sharing	normal	572.16
		dry	286.08
2-Basra	us_priority	normal	228.15
		dry	0
	prop_sharing	normal	228.15
		dry	114.075

Table 09: Shadow Price of Water by Province, Crop, Shortage Sharing Arrangement, and Water Supply Scenario, Tigris and Euphrates Basins, Iraq, 2006, ($US per 1000 cubic meters)

1-Mousil	us_priority	normal	0.00
		dry	9.47
	prop_sharing	normal	0.00
		dry	9.47
2-Basra	us_priority	normal	0.00
		dry	10.83
	prop_sharing	normal	0.00
		dry	10.83

Model 6

Code

```
$EOLCOM //
$TITLE Iraq Water Rights Priority Model
$OFFSYMXREF OFFSYMLIST ONLISTING OFFUPPER

OPTION LIMROW=000, LIMCOL = 0;

$ONTEXT

* ----------------------------------------------------------------------------------------
  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  Dina A. Salman:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, New Mexico USA
  email: dinasalm@nmsu.edu

  Saud A.Amer:
  US Geological Survey - International Water Resources Division
  Reston VA  USA
  email:  samer@usgs.gov

  May 13, 2013

* ----------------------------------------------------------------------------------------
*  Model 6 Simple Farm Management
*  Expands scope and scale of Model 5
*  2 to 13 provinces
*  2 to 3 water supply scenarios
*  2 to 4 water shortage sharing methods
*  2 to 9 crops
* ----------------------------------------------------------------------------------------

* model 6, adds 13 provinces and 9 crops and 4 water sharing arrangements


$OFFTEXT

****************************************   Section 1 ***********************************
*                                           Sets                                        *
```

```
*****************************************************************************************

SETS

*****************************************************************************************
i      province
*****************************************************************************************

/       1-Mousil
        2-Kurkuk
        3-Salaheldeen
        4-Deyala
        5-Anbar
        6-Baghdad
        7-Babylon
        8-Karbala
        9-Qadeseeya
       10-Muthana
       11-Thieqar
       12-Meesan
       13-Basra
/


k     crop

/ 1-Rice
  2-Wheat
  3-Cotton
  4-Sunflower
  5-Maize
  6-Barley
  7-Tomato
  8-Lettuce
  9-Onion
/


s   hydrologic water supply scenario

/
1-normal
```

```
2-dry
3-drought /


j   water right priority:  water right priority: same number # of elements as # of provinces - 1 priority per province


/      j1*j13
/

set r water allocation rule allows for many water shortage sharing rules to be considered

/1-ds_priority   downstream priority                                                  // rule 1
 2-us_priority   upstream   priority                                                  // rule 2
 3-prop_sharing  proportional sharing of shortages                                    // rule 3
 4-free_market   free market - proportional sharing then water moves by market        // rule 4
/

*subsets of r
set rwa(r) / 1-ds_priority, 2-us_priority, 3-prop_sharing/   // subset of rules = everything except market trading
set rfm(r) / 4-free_market/                                  // subset of rules = only market trading of water in drought


Set rji(r, j, i) mapping set: assigns priorities to province - separate assignments for each proposed rule

// best (senior) right assigned to canal with j1 priority - worst (junior) priority assigned to canal with j4
// Two canals with equal priority (e.g. two both with j1) share shortages proportionally

/1-ds_priority  . (j1.13-Basra,  j2. 12-Meesan,  j3.11-Thieqar,    j4.10-Muthana, j5.9-Qadeseeya, j6.8-Karbala,  j7.7-Babylon,  j8.6-
Baghdad,  j9.5-Anbar,     j10.4-Deyala,   j11.3-Salaheldeen, j12.2-Kurkuk,  j13.1-Mousil)
 2-us_priority  . (j1.1-Mousil,  j2.2-Kurkuk,    j3.3-Salaheldeen, j4.4-Deyala,   j5.5-Anbar,     j6.6-Baghdad,  j7.7-Babylon,  j8.8-
Karbala,  j9.9-Qadeseeya, j10.10-Muthana, j11.11-Thieqar,    j12.12-Meesan, j13.13-Basra )
 3-prop_sharing . j1. (1-Mousil, 2-Kurkuk, 3-Salaheldeen, 4-Deyala, 5-Anbar, 6-Baghdad, 7-Babylon, 8-Karbala, 9-Qadeseeya, 10-
Muthana, 11-Thieqar, 12-Meesan, 13-Basra )
 4-free_market  . j1. (1-Mousil, 2-Kurkuk, 3-Salaheldeen, 4-Deyala, 5-Anbar, 6-Baghdad, 7-Babylon, 8-Karbala, 9-Qadeseeya, 10-
Muthana, 11-Thieqar, 12-Meesan, 13-Basra )
/

************************************* Section 2 ***************************************************************************
*                                        Data                                                                              *
****************************************************************************************************************************
```

```
TABLE Bc(i, k)    ET - Per hectare crop water demand 10ths of meters depth  = 1000s cubic meters per ha.

* Data:  Allen, R. G., L.S. Pereira, D. Raes, and M. Smith, 1998. Crop Evapotranspiration Guidelines for Computing
* Crop Water Requirements: Food and Agricultural Organization Report 56.

* ----- Column Heads are Crops -- Row heads are irrigation canals (water use areas) -----

                     1-Rice        2-wheat     3-cotton  4-sunflower  5-Maize   6-Barley     7-tomato     8-lettuce     9-onion
*-------------------------- use node rows ------------------------------------------------------------------------------------
 1-Mousil              30.8         11.9        18.0        12.9         7.0        2.8         7.1          1.7          9.5
 2-Kurkuk              32.4         12.3        19.1        13.4         7.8        2.6         7.9          1.6         10.0
 3-Salaheldeen         28.4         9.9         16.6        10.9         6.8        2.5         6.9          1.6          9.4
 4-Deyala              29.6         10.8        17.3        11.8         6.6        3.1         6.7          2.4         10.1
 5-Anbar               33.8         12.4        19.7        13.6         8.8        3.1         8.8          1.9         11.7
 6-Baghdad             32.2         11.8        18.8        13.0         8.2        2.8         8.2          1.7         10.8
 7-Babylon             32.5         11.9        18.6        13.0         8.2        3.0         8.2          1.8         10.9
 8-Karbala             32.8         11.9        18.8        13.1         8.6        3.1         8.6          1.9         11.5
 9-Qadeseeya           34.2         12.3        19.6        13.6         9.2        3.4         9.2          2.0         12.3
 10-Muthana            34.4         12.3        19.6        13.5         9.3        3.6         9.3          2.1         12.8
 11-Thieqar            35.6         12.8        20.3        14.1         9.0        3.0         9.1          1.9         11.9
 12-Meesan             34.8         12.5        20.1        13.8         9.2        3.4         9.2          2.1         12.3
 13-Basra              37.2         13.4        21.4        14.9         9.7        3.4         9.8          2.1         12.7
*-----------------------------------------------------------------------------------------------------------------------------
;

*Bc(k) =  0.5 * Bc(k)

Table    Yield_p(i,k)   Crop Yield tons per Ha (proportional to ET)

                    1-Rice        2-wheat    3-cotton  4-sunflower   5-Maize     6-Barley    7-tomato    8-lettuce        9-onion
*-------------------------- use node rows (+) ---------------------------------------------------------------------------------
  1-Mousil          2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  2-Kurkuk          2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  3-Salaheldeen     2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  4-Deyala          2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  5-Anbar           2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  6-Baghdad         2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  7-Babylon         2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  8-Karbala         2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  9-Qadeseeya       2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
  10-Muthana        2.89         1.38        2.42        1.33          2.43        0.90        15.8        20.7          7.40
```

```
  11-Thieqar        2.89         1.38         2.42         1.33          2.43         0.90        15.8          20.7           7.40
  12-Meesan         2.89         1.38         2.42         1.33          2.43         0.90        15.8          20.7           7.40
  13-Basra          2.89         1.38         2.42         1.33          2.43         0.90        15.8          20.7           7.40
*-----------------------------------------------------------------------------------------------------------------------------
;

* use formula below to reduce yields with cooler climates and increase them with warmer .

Yield_p(i,k) = yield_p ('6-baghdad',k) *  Bc(i, k)/ Bc('6-baghdad', k) ;


* economic data

Parameter   Price_p(k)        Crop Prices ($ US per ton)

/1-Rice                      985
 2-wheat                     225
 3-cotton                    906
 4-sunflower                 415
 5-Maize                     180
 6-Barley                    175
 7-tomato                    115
 8-lettuce                   120
 9-onion                     142
/
;


Table  Cost_p(i,k)     Crop Production Costs Excluding water ($ US per Ha)

                  1-Rice      2-wheat    3-cotton  4-sunflower    5-Maize   6-Barley  7-tomato    8-lettuce      9-onion
*------------------------- use node rows (+) -------------------------------------------------------------------------

 1-Mousil            180        200         1300          220         70         100       60          80             70
 2-Kurkuk            180        200         1300          220         70         100       60          80             70
 3-Salaheldeen       180        200         1300          220         70         100       60          80             70
 4-Deyala            180        200         1300          220         70         100       60          80             70
 5-Anbar             180        200         1300          220         70         100       60          80             70
 6-Baghdad           180        200         1300          220         70         100       60          80             70
 7-Babylon           180        200         1300          220         70         100       60          80             70
 8-Karbala           180        200         1300          220         70         100       60          80             70
 9-Qadeseeya         180        200         1300          220         70         100       60          80             70
```

```
 10-Muthana          180       200       1300        220        70        100       60        80          70
 11-Thieqar          180       200       1300        220        70        100       60        80          70
 12-Meesan           180       200       1300        220        70        100       60        80          70
 13-Basra            180       200       1300        220        70        100       60        80          70
*------------------------------------------------------------------------------------------------------------------------
;

* 1/4 of all costs relate to yield, and 3/4 are independent of yield

parameter cost_harvest_p(i,k)  production costs per ha related to crop yield
;

cost_harvest_p(i,k) = 0.25 * cost_p(i,k);  // harvest-related costs approx 25% of all prodn costs - data source needed
cost_harvest_p(i,k) =  cost_harvest_p('6-Baghdad', k) * yield_p(i,k) / yield_p('6-baghdad', k);  // costs related to harvests


parameter cost_n_harvest_p(i,k)  costs not related to (independent of) harvest;

cost_n_harvest_p(i,k) = 0.75 * cost_p(i,k)  // higher total prodn costs with higher yields.  Baghdad is reference value

* re assemble cost below.  For any given crop we have (1) rising costs with higher yields and (2)  constant costs over provinces

parameter cost_p(i,k)  total cost per ha;

Cost_p(i,k)= Cost_harvest_p(i,k) + cost_n_harvest_p(i,k);


Table land_p(i,  k)   observed land in prodn (1000 Ha)

                    1-Rice          2-wheat      3-cotton  4-sunflower      5-Maize    6-Barley    7-tomato    8-lettuce    9-onion
*-------------------------- use node rows (+) -------------------------------------------------------------------------
   1-Mousil           0.0             47.4         0.45        0.4            0.0        77.1        43.6         3.7          0.0
   2-Kurkuk           0.0             26.1         1.60        0.0            0.5         1.5         6.9         2.0          0.0
   3-Salaheldeen      0.0            145.5         3.00        6.4           40.3        13.6        46.3        69.1          7.2
   4-Deyala           0.6            107.1         1.40        0.3           10.3        22.2        17.7        12.9          6.2
   5-Anbar            0.2             55.5         0.15        2.2            7.2         3.4        13.2        13.6         27.6
   6-Baghdad          0.0             65.7         2.60        2.0           25.6         8.2        51.8        27.6         13.7
   7-Babylon          1.6             74.2         1.00        0.6           73.0        44.4        15.6        15.3         47.0
   8-Karbala          0.0              2.9         0.04        0.2            3.1         2.9        11.8         7.8          4.8
   9-Qadeseeya       32.5             91.8         0.40        0.05           3.8        72.3        15.2         9.6         15.3
   10-Muthana         4.5             12.8         0.00        0.0            0.0        29.1         1.6         0.7         10.3
   11-Thieqar         0.1             61.3         0.05        0.03           3.4        81.5        16.7        11.1         11.4
```

```
   12-Meesan          6.8               81.0         0.00        0.005          12.1         65.5       18.6          8.9        20.0
   13-Basra           0.0               16.9         0.00        0.0            8.6           0.0       10.5         24.9         0.1
;

Parameter Net_revenue_p  (i,k)  net revenue per ha observed
*          T_net_revenue_p(i  )  total net revenue by province summed over crops
;

Net_revenue_p  (i,k) = Price_p(k) * Yield_p(i,k) - Cost_p(i,k)         + eps;

parameter Wat_supply_p(s)  total water available (million cubic meters per year) calculated total ag water use (not meas from gauges)
;

Wat_supply_p('1-normal')  =  sum((i,k), Bc(i,k) * land_p(i,k));    // actual water use in full supply year
Wat_supply_p('2-dry')     =  0.50 * Wat_supply_p('1-normal');              // 50% of full supply
Wat_supply_p('3-drought') =  0.20 * Wat_supply_p('1-normal');              // 20% of full supply

parameter right_p(i)  total water assigned by province in full supply conditions (paper water)
;

right_p(i)  =  {sum(k, Bc(i,k) * land_p(i,k))};

*******************  code defines water sharing methods *****************************************

* code below loops over each priority (j) then maps each priority to corresponding province i
* code is thanks to Pete Stacy at GAMS Development Corporation Feb 1 2012

parameter   tot_assigned(r,s) cumulative assignment at canal i including higher priorities;
            tot_assigned(r,s)  = 0;   // starts at 0 - prepares to loop

parameter
  tot_pap_right_by_prior(  r,j) total paper rights by jth priority for all canals including ties
  remain_supply_by_prior(r,s,j) residual supply    by jth priority after supplying higher priorities
  wet_wat_use           (r,s,i) wet water use assigned to ith limited by total basin supply
;

Loop(r,                                                                            // loop over sharing rule (r)
 Loop(j,                                                                           // loop over priority (j)

   tot_pap_right_by_prior(r,j) = sum[i$rji(r, j, i), right_p(i)] ;   // total paper rights by (j) after protecting higher priorities
   remain_supply_by_prior(r,s,j) =
    min[(Wat_supply_p(s) - tot_assigned(r,s)), tot_pap_right_by_prior(r,j)] + eps;  // remaining supply by jth priority after
```

```
*supplying higher priorities
   Loop(i$rji(r, j, i),
       wet_wat_use(r,s,i) = (right_p(i)/tot_pap_right_by_prior(r,j))
       * remain_supply_by_prior(r,s,j) + eps;                        // wet water assigned to (i) province
    tot_assigned(r,s) = tot_assigned(r,s) + wet_wat_use(r,s,i) ;      // cumulative water (check) assigned to last province
*getting water - should match total supply

   );                                                               // end province loop
  );                                                                // end priority loop
);                                                                  // end rule loop



parameter tot_wat_use(r,s)  total water use;
tot_wat_use  (r,s) = sum(i, wet_wat_use(r,s,i)) + eps;

************** end of water allocation system ******************************************

**** begin looping over water sharing rule and water supply scenario below
* rr, ss, are subset of orginal sets... allows fast scenario analysis below

set rr(r);    // water sharing rule
set ss(s);    // water supply scenario

rr(r) = no;   // switch rr subset off -- prepares to turn it on below in equations
ss(s) = no;   // switch ss subset off -- prepares to turn it on below in equations

******************************  SECTION 3 ***********************************************
*                               Variables                                              *
*****************************************************************************************

POSITIVE VARIABLES

hectares_v                (r,s,i,k)    land in production by rule-province-crop-scen    (1000 Ha - marginal is $US per Ha)
T_hectares_v              (r,s,i  )    total land in prodn by rule-province-scen        (1000 Ha - marginal is $US per Ha)

Uses_v                    (r,s,i  )    total water use by rule-province-crop-scen       (million m^3 - marginal is $US per 1000 m^3)
Uses_Crop_v               (r,s,i,k)    total water use by rule-province-scen            (million m^3 - marginal is $US per 1000 m^3)

Sum_uses_v                (r,s    )    summed wat uses over provinces by rule-scen      (million m^3 - marginal is $US per 1000 m^3)

variables
```

```
Ag_Ben_k_v                (r,s,i,k)    total farm income by rule-scenario-province-crop ($US 1000s - no marginals shown)
Ag_Ben_v                  (r,s    )    total farm income by rule-scenario-province      ($US 1000s - no marginals shown)

Tot_b_v        total farm income re-calc for each rule-scen (objective)                 ($US 1000s - no marginals shown)

;
*******************************   Section 4  ******************************************
*                                 Equations                                            *
****************************************************************************************


EQUATIONS

// Equations DECLARED

T_hectares_e             (r,s,i      )      total land in production by province-scen  (1000 Ha - marginal is $US per Ha)

Uses_crop_e              (r,s,i,k    )      total water use by rule-scen-prov-crop     (million m^3 - marginal is $US per 1000 m^3)
Uses_e                   (r,s,i      )      total water use by rule-scen-prov          (million m^3 - marginal is $US per 1000 m^3)

Sum_uses_e               (r,s        )      total water use by rule-scen               (million cubic meters per year)

ag_ben_k_e               (r,s,i,   k )      total farm income by rule-scen-prov-crop   ($US 1000)
Ag_ben_e                 (r,s        )      total farm income by rule-scen-prov        ($US 1000)

Tot_b_e                                     total farm income by element of loop       ($US 1000)
;


// Equations defined (using above names with algebraic formulas)

// Equations below defined over rule (rr) and scenario (ss) using subset rr and ss

T_hectares_e(rr,ss,i  )..  T_hectares_v(rr,ss,i  ) =e=      sum(k, hectares_v(rr,ss,i,k));

Uses_crop_e (rr,ss,i,k)..  Uses_crop_v(rr,ss,i,k)  =e=   Bc(i,k) * hectares_v(rr,ss,i,k) ;
Uses_e      (rr,ss,i  )..       Uses_v(rr,ss,i  )  =e=     sum(k, Uses_crop_v(rr,ss,i,k));
sum_uses_e  (rr,ss    )..   sum_uses_v(rr,ss    )  =e=          sum(i, Uses_v(rr,ss,i  ));

ag_ben_k_e  (rr,ss,i,k)..  Ag_Ben_k_v (rr,ss,i,k)  =e=   Net_revenue_p(i,k) * hectares_v(rr,ss,i,k);
Ag_ben_e    (rr,ss    )..  Ag_Ben_v   (rr,ss    )  =e=    sum((i,k), Ag_Ben_k_v(rr,ss,i,k));
```

```
Tot_b_e                ..  Tot_b_v     =e= sum((rr,ss), ag_ben_v(rr,ss    ));  //sums total benefits over indices for each rule-scen

**************** Section 5 **************************************************************
*  Labels and defines models                                                              *
*  Each model has one objective function                                                  *
*****************************************************************************************

MODEL TE_06 /all/;

**********************************  Section 6   ****************************************
*                                   BOUNDS                                                *
*  Bounding variables lets you see a non-zero shadow price if a resource is limiting    *
*****************************************************************************************

uses_v.up   (rwa,s, i)  = wet_wat_use(rwa,s,i);  // without water trading water use limited to historic by province
sum_uses_v.up(rfm,s  )  = tot_wat_use(rfm,s  );  // with water trading only total (basin) water is bounded

hectares_v.up(r,s,i,k)  = land_p(i,k);           // can produce no more crops than observed under full water supply

**************** Section 7 **************************************************************
*               MODEL SOLVES                                                              *
*****************************************************************************************

parameter shad_price_p(r,s,i)  shadow prices calculated below after each optimization
;

* prepare for multiple solves inside the loops

  loop(r,
   loop(s,

     ss(s) = yes;
     rr(r) = yes;

     Solve TE_06 using nlp maximizing Tot_b_v;

     shad_price_p(rwa,s,i    ) =            uses_v.m(rwa,s,i) + eps;   // shad prices for all water allocation rules ex trading
     shad_price_p(rfm,s,i    ) =        sum_uses_v.m(rfm,s  ) + eps;   // shad prices for trading -  should be = across provinces

* closes loops below over r,s (rule for sharing water shortages, water supply scenario)
```

```
    rr(r) = no;
    ss(s) = no;

   );
  );

*****************************************************************************************
*   Section 8: DISPLAYS                                                                 *
*****************************************************************************************

*  Parameters are calculated values from the optimal solution, first defined then found

parameter

tot_use_p           (r,s,i,k  )   total water use by rule-scen-prov-crop
uses_p              (r,s,i    )   total water use by rule-scen_prov

land_v_p            (r,s,i,k  )   total land in production

ben_by_crop_canal_p (r,s,i,k  )   total benefits by crop and province
ben_by_canal_p      (r,s,i    )   total economic benefits by province
tot_ben_p           (r,s      )   total economic benefits
;

* land in prodn (1000 hectares)

land_v_p            (r,s,i,k) =    hectares_v.l(r,s,i,k)  + eps;

* water (1000 hectare meters)

tot_use_p           (r,s,i,k) =   Uses_crop_v.l(r,s,i,k)  + eps;
uses_p              (r,s,i  ) =        Uses_v.l(r,s,i  )  + eps;

* economic benefits ($1000 US)

ben_by_crop_canal_p (r,s,i,k) =    Ag_ben_k_v.l    (r,s,i,k)  + eps;
ben_by_canal_p      (r,s,i  ) = sum(k, ag_ben_k_v.l(r,s,i,k)) + eps;
tot_ben_p           (r,s    ) =    ag_ben_v.l      (r,s    )  + eps;


* GAMS GDX facility writes to external spreadsheet
```

```
execute_unload "farm_mgmt_06.gdx"

tot_pap_right_by_prior,        remain_supply_by_prior,  wet_wat_use,   tot_assigned
land_p,          cost_p,         land_v_p,    tot_use_p,  uses_p,        ben_by_crop_canal_p,
ben_by_canal_p, tot_ben_p,      net_revenue_p, shad_price_p, Price_p
yield_p,         Net_revenue_p, Bc
;

$onecho > gdxxrwout.txt

i=farm_mgmt_06.gdx
o=farm_mgmt_06.xls

epsout = 0

* cdim = 0 puts output into LIST format, one piece of data per line

par = net_revenue_p          rng =  data_net_rev_per_ha!c4             Cdim = 0

par = tot_pap_right_by_prior rng =  total_paper_right_by_priority!c4   Cdim = 0
par = remain_supply_by_prior rng =  remain_supply_by_priority!c4       cdim = 0
par = wet_wat_use            rng =  wet_water_use_by_province!c4       cdim = 0
par = tot_assigned           rng =  basin_cum_water_use!c4             cdim = 0
par = land_v_p               rng =  land_in_prodn!c4                   cdim = 0
par = tot_use_p              rng =  water_use_by_crop_canal!c4         cdim = 0
par = uses_p                 rng =  total_water_use_by_canal!c4        cdim = 0
par = ben_by_crop_canal_p    rng =  benefits_by_crop_canal!c4          cdim = 0

par = Bc                     rng =  data_wu_ha!c4                      cdim = 0

par = ben_by_canal_p         rng =  benefits_by_canal!c4               cdim = 0
par = tot_ben_p              rng =  total_benefits!c4                  cdim = 0
par = shad_price_p           rng =  shadow_price!c4                    cdim = 0

par = Price_p                rng =  data_price!c4                      cdim = 0
par = cost_p                 rng =  data_cost_per_ha!c4                cdim = 0
par = yield_p                rng =  data_yield!c4                      cdim = 0

$offecho
execute 'gdxxrw.exe @gdxxrwout.txt trace=2';

// THE END
```

Model 6

Tabled Results

Table 1: Crop Coefficient by Crop and Growth Stage, Tigris-Euphrates Basin, Iraq

Crops	Days to maturity	Planting Month	Growth Stage							
			1		2		3		4	
			Days in stage	Crop Coefficient	Days in stage	Crop Coefficient	days in stage	Crop Coefficient	days in stage	Crop Coefficient
Rice	180	May	30	1.05	30	1.10	80	1.20	40	0.90
Wheat	130	April	40	0.00	30	0.25	40	1.15	20	0.35
Cotton	180	April	30	0.00	50	0.25	55	1.15	45	0.60
Sunflower	130	April	25	0.00	35	0.25	45	1.15	25	0.35
Maize	140	January	25	0.00	40	0.25	45	1.20	30	0.50
Barley	120	November	15	0.00	25	0.25	50	1.15	30	0.25
Tomato	135	January	30	0.00	40	0.25	40	1.15	25	0.80
Lettuce	100	October	25	0.00	35	0.10	30	1.00	10	0.95
Onion	210	October	20	0.00	35	0.25	110	1.05	45	0.95

Table 2: Crop Evapotranspiration (ET) by Crop and Province, Lower Tigris-Euphrates Basin, Iraq, 2006 (mm depth per year)[1]									
Province	Crops								
	Rice	Wheat	Cotton	Sunflower	Maize	Barley	Tomato	Lettuce	Onion
1-Mousil	3,080	1,190	1,804	1,292	702	291	706	168	952
2-Kurkuk	3,236	1,228	1,914	1,342	781	258	792	163	997
3-Salaheldeen	2,839	991	1,663	1,095	683	254	688	156	939
4-Deyala	2,962	1,082	1,732	1,181	662	307	669	244	1,013
5-Anbar	3,377	1,236	1,973	1,358	881	310	875	186	1,168
6-Baghdad	3,220	1,181	1,881	1,302	821	284	822	173	1,084
7-Babylon	3,248	1,187	1,861	1,301	818	299	825	180	1,094
8-Karbala	3,281	1,189	1,878	1,311	859	311	864	185	1,148
9-Qadeseeya	3,419	1,232	1,955	1,357	918	335	920	199	1,235
10-Muthana	3,438	1,226	1,958	1,351	934	358	934	213	1,282
11-Thieqar	3,558	1,277	2,031	1,406	903	305	908	194	1,186
12-Meesan	3,482	1,254	2,013	1,384	918	344	921	211	1,229
13-Basra	3,723	1,345	2,141	1,490	970	344	977	211	1,270

[1] Source for Data and Method (Allen, et al, 1998). ET was calculated as the product of ET for a base crop (ET_0) and a crop coefficient unique to each crop (Kc)

Table 3: Average Price by Crop, Lower Tigris-Euphrates, Iraq, 2006 ($US / ton)	
Crop	Price
1-Rice	985
2-Wheat	225
3-Cotton	906
4-Sunflower	415
5-Maize	180
6-Barley	175
7-Tomato	115
8-Lettuce	120
9-Onion	142

Table 4: Baseline Crop Yield by Province and Crop, Lower Tigris-Euphrates Basin, Iraq, 2006 (metric tons/ha)									
	Crop								
Province	1-Rice	2-Wheat	3-Cotton	4-Sunflower	5-Maize	6-Barley	7-Tomato	8-Lettuce	9-Onion
1-Mousil	2.76	1.39	2.32	1.32	2.07	0.90	13.68	20.70	6.51
2-Kurkuk	2.91	1.44	2.46	1.37	2.31	0.84	15.22	19.48	6.85
3-Salaheldeen	2.55	1.16	2.14	1.12	2.02	0.80	13.30	19.48	6.44
4-Deyala	2.66	1.26	2.23	1.21	1.96	1.00	12.91	29.22	6.92
5-Anbar	3.03	1.45	2.54	1.39	2.61	1.00	16.96	23.14	8.02
6-Baghdad	2.89	1.38	2.42	1.33	2.43	0.90	15.80	20.70	7.40
7-Babylon	2.92	1.39	2.39	1.33	2.43	0.96	15.80	21.92	7.47
8-Karbala	2.94	1.39	2.42	1.34	2.55	1.00	16.57	23.14	7.88
9-Qadeseeya	3.07	1.44	2.52	1.39	2.73	1.09	17.73	24.35	8.43
10-Muthana	3.09	1.44	2.52	1.38	2.76	1.16	17.92	25.57	8.77
11-Thieqar	3.20	1.50	2.61	1.44	2.67	0.96	17.53	23.14	8.15
12-Meesan	3.12	1.46	2.59	1.41	2.73	1.09	17.73	25.57	8.43
13-Basra	3.34	1.57	2.75	1.52	2.87	1.09	18.88	25.57	8.70

Table 5: Production Cost per Ha, by Province and Crop, Lower Tigris-Euphrates Basin, Iraq, 2006 ($US per ha)

	Crop								
province	1-Rice	2-Wheat	3-Cotton	4-Sunflower	5-Maize	6-Barley	7-Tomato	8-Lettuce	9-Onion
1-Mousil	178	200	1286	220	67	100	58	80	68
2-Kurkuk	180	202	1305	222	69	98	59	79	69
3-Salaheldeen	175	192	1262	211	67	97	58	79	68
4-Deyala	176	196	1274	215	67	103	57	88	69
5-Anbar	182	203	1316	223	71	103	61	82	71
6-Baghdad	180	200	1300	220	70	100	60	80	70
7-Babylon	180	200	1297	220	70	102	60	81	70
8-Karbala	181	200	1300	220	71	103	61	82	71
9-Qadeseeya	183	202	1314	223	72	105	62	84	72
10-Muthana	183	202	1314	222	72	107	62	85	73
11-Thieqar	185	204	1326	225	72	102	62	82	72
12-Meesan	184	203	1322	223	72	105	62	85	72
13-Basra	187	207	1345	228	73	105	63	85	73

Table 6: Average Net Farm Income per Ha by Crop and Province, Lower Tigris-Euphrates Basin, Iraq, 2006 base conditions ($US per ha)									
	crop								
province	1-Rice	2-Wheat	3-Cotton	4-Sunflower	5-Maize	6-Barley	7-Tomato	8-Lettuce	9-Onion
1-Mousil	2,545	113	813	328	306	58	1,515	2,404	856
2-Kurkuk	2,684	122	922	347	347	48	1,691	2,259	904
3-Salaheldeen	2,336	69	674	252	296	43	1,471	2,259	847
4-Deyala	2,440	88	744	286	285	72	1,427	3,419	914
5-Anbar	2,806	124	982	355	398	72	1,889	2,694	1,067
6-Baghdad	2,667	111	893	332	367	58	1,757	2,404	981
7-Babylon	2,693	113	873	332	367	67	1,757	2,549	990
8-Karbala	2,719	113	893	336	388	72	1,845	2,694	1,048
9-Qadeseeya	2,841	122	972	355	419	86	1,977	2,839	1,124
10-Muthana	2,858	122	972	351	424	95	1,999	2,984	1,172
11-Thieqar	2,962	133	1,042	374	408	67	1,955	2,694	1,086
12-Meesan	2,893	126	1,022	363	419	86	1,977	2,984	1,124
13-Basra	3,102	146	1,151	405	444	86	2,109	2,984	1,163

Table 7: Regional Farm Income by Water Sharing Arrangement, Water Supply, and Province, Lower Tigris-Euphrates Basin, Iraq, 2006 ($US million/year)

	Priority											
	1-ds_priority			2-us_priority			3-prop_sharing			4-free_market		
supply / Province	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought
1-Mousil	85.2	0.0	0.0	85.2	85.2	85.2	85.2	80.0	54.9	85.2	79.9	75.0
2-Kurkuk	21.1	0.0	0.0	21.1	21.1	21.1	21.1	19.0	17.4	21.1	17.9	16.2
3-Salaheldeen	256.4	0.0	0.0	256.4	256.4	254.9	256.4	248.2	228.1	256.4	246.5	224.2
4-Deyala	91.6	0.0	0.0	91.6	91.6	0.0	91.6	85.3	80.0	91.6	82.2	69.4
5-Anbar	102.5	0.0	0.0	102.5	102.5	0.0	102.5	96.2	71.7	102.5	95.6	91.0
6-Baghdad	190.9	0.0	0.0	190.9	190.9	0.0	190.9	181.3	129.3	190.9	183.7	157.4
7-Babylon	156.5	149.4	0.0	156.5	118.5	0.0	156.5	137.9	95.2	156.5	148.1	71.0
8-Karbala	49.7	49.7	0.0	49.7	0.0	0.0	49.7	43.3	28.3	49.7	49.3	47.8
9-Qadeseeya	186.2	186.2	0.0	186.2	0.0	0.0	186.2	165.1	93.5	186.2	175.0	74.5
10-Muthana	34.6	34.6	0.0	34.6	0.0	0.0	34.6	28.5	14.1	34.6	33.0	17.4
11-Thieqar	90.3	90.3	88.2	90.3	0.0	0.0	90.3	83.1	71.8	90.3	82.1	74.9
12-Meesan	126.4	126.4	126.4	126.4	0.0	0.0	126.4	116.2	82.9	126.4	116.2	85.8
13-Basra	102.8	102.8	102.8	102.8	0.0	0.0	102.8	100.1	83.1	102.8	100.5	96.6
Total	1,494.1	739.3	317.5	1,494.1	866.3	361.2	1,494.1	1,384.4	1,050.3	1,494.1	1,410.0	1,101.1

Table 7: Regional Farm Income by Water Sharing Arrangement, Water Supply, and Province, Lower Tigris-Euphrates Basin, Iraq, 2006 ($US million/year)

	Priority											
	1-ds_priority			2-us_priority			3-prop_sharing			4-free_market		
supply / Province	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought
1-Mousil	85.2	0.0	0.0	85.2	85.2	85.2	85.2	80.0	54.9	85.2	79.9	75.0
2-Kurkuk	21.1	0.0	0.0	21.1	21.1	21.1	21.1	19.0	17.4	21.1	17.9	16.2
3-Salaheldeen	256.4	0.0	0.0	256.4	256.4	254.9	256.4	248.2	228.1	256.4	246.5	224.2
4-Deyala	91.6	0.0	0.0	91.6	91.6	0.0	91.6	85.3	80.0	91.6	82.2	69.4
5-Anbar	102.5	0.0	0.0	102.5	102.5	0.0	102.5	96.2	71.7	102.5	95.6	91.0
6-Baghdad	190.9	0.0	0.0	190.9	190.9	0.0	190.9	181.3	129.3	190.9	183.7	157.4
7-Babylon	156.5	149.4	0.0	156.5	118.5	0.0	156.5	137.9	95.2	156.5	148.1	71.0
8-Karbala	49.7	49.7	0.0	49.7	0.0	0.0	49.7	43.3	28.3	49.7	49.3	47.8
9-Qadeseeya	186.2	186.2	0.0	186.2	0.0	0.0	186.2	165.1	93.5	186.2	175.0	74.5
10-Muthana	34.6	34.6	0.0	34.6	0.0	0.0	34.6	28.5	14.1	34.6	33.0	17.4
11-Thieqar	90.3	90.3	88.2	90.3	0.0	0.0	90.3	83.1	71.8	90.3	82.1	74.9
12-Meesan	126.4	126.4	126.4	126.4	0.0	0.0	126.4	116.2	82.9	126.4	116.2	85.8
13-Basra	102.8	102.8	102.8	102.8	0.0	0.0	102.8	100.1	83.1	102.8	100.5	96.6
Total	1,494.1	739.3	317.5	1,494.1	866.3	361.2	1,494.1	1,384.4	1,050.3	1,494.1	1,410.0	1,101.1

Table 8: Irrigated Land in Production, by Province, Crop, Shortage Sharing Arrangement, and Water Supply Scenario, Lower Tigris-Euphrates Basins, Iraq, 2006, (1000 hectares/year)

	Priority											
	1-ds_priority			2-us_priority			3-prop_sharing			4-free_market		
Supply / Province	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought
1-Mousil	173	0	0	173	173	173	173	126	34	173	125	47
2-Kurkuk	39	0	0	39	39	39	39	22	10	39	13	9
3-Salaheldeen	331	0	0	331	331	308	331	212	120	331	186	115
4-Deyala	179	0	0	179	179	0	179	107	48	179	72	31
5-Anbar	123	0	0	123	123	0	123	72	36	123	68	54
6-Baghdad	197	0	0	197	197	0	197	118	63	197	132	79
7-Babylon	273	210	0	273	82	0	273	134	60	273	199	36
8-Karbala	34	34	0	34	0	0	34	20	12	34	31	24
9-Qadeseeya	241	241	0	241	0	0	241	72	47	241	149	40
10-Muthana	59	59	0	59	0	0	59	17	10	59	46	13
11-Thieqar	186	186	170	186	0	0	186	132	36	186	124	39
12-Meesan	213	213	213	213	0	0	213	132	45	213	132	48
13-Basra	61	61	61	61	0	0	61	43	29	61	45	36
Total	2107	1003	444	2107	1123	520	2107	1207	551	2107	1320	571

Table 9: Total Water Use Summed over Crops, by Province, Crop, Shortage Sharing Arrangement, and Water Supply Scenario, Lower Tigris-Euphrates Basins, Iraq, 2006, (million cubic meters/year)

	Priority											
	1-ds_priority			2-us_priority			3-prop_sharing			4-free_market		
Supply	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought
Province												
1-Mousil	1,109	0	0	1,109	1,109	1,109	1,109	555	222	1,109	545	316
2-Kurkuk	417	0	0	417	417	417	417	209	83	417	96	58
3-Salaheldeen	2,366	0	0	2,366	2,366	2,138	2,366	1,183	473	2,366	925	430
4-Deyala	1,551	0	0	1,551	1,551	0	1,551	776	310	1,551	394	150
5-Anbar	1,267	0	0	1,267	1,267	0	1,267	633	253	1,267	578	465
6-Baghdad	1,703	0	0	1,703	1,703	0	1,703	851	341	1,703	927	472
7-Babylon	2,361	1,614	0	2,361	747	0	2,361	1,180	472	2,361	1,478	206
8-Karbala	245	245	0	245	0	0	245	123	49	245	211	172
9-Qadeseeya	2,877	2,877	0	2,877	0	0	2,877	1,439	575	2,877	1,748	347
10-Muthana	565	565	0	565	0	0	565	283	113	565	408	148
11-Thieqar	1,373	1,373	1,179	1,373	0	0	1,373	687	275	1,373	589	309
12-Meesan	2,019	2,019	2,019	2,019	0	0	2,019	1,010	404	2,019	1,007	436
13-Basra	466	466	466	466	0	0	466	233	93	466	253	156
Total	18,320	9,160	3,664	18,320	9,160	3,664	18,320	9,160	3,664	18,320	9,160	3,664

Table 10: Shadow Price of Water by Province, Crop, Shortage Sharing Arrangement, and Water Supply Scenario, Lower Tigris-Euphrates Basins, Iraq, 2006, ($US per 1000 cubic meters)

	Priority											
	1-ds_priority			2-us_priority			3-prop_sharing			4-free_market		
supply	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought	1-normal	2-dry	3-drought
province												
1-Mousil	0	1414	1414	0	0	0	0	9	213	0	11	91
2-Kurkuk	0	1412	1412	0	0	0	0	10	48	0	11	91
3-Salaheldeen	0	1412	1412	0	0	7	0	7	90	0	11	91
4-Deyala	0	1424	1424	0	0	1424	0	8	43	0	11	91
5-Anbar	0	1418	1418	0	0	1418	0	10	91	0	11	91
6-Baghdad	0	1414	1414	0	0	1414	0	45	214	0	11	91
7-Babylon	0	9	1416	0	45	1416	0	45	91	0	11	91
8-Karbala	0	0	1418	0	1418	1418	0	91	215	0	11	91
9-Qadeseeya	0	0	1419	0	1419	1419	0	83	83	0	11	91
10-Muthana	0	0	1421	0	1421	1421	0	83	92	0	11	91
11-Thieqar	0	0	10	0	1418	1418	0	10	91	0	11	91
12-Meesan	0	0	0	0	1421	1421	0	10	91	0	11	91
13-Basra	0	0	0	0	1421	1421	0	46	215	0	11	91
Average	0	654	1091	0	659	1092	0	35	121	0	11	91

Article from Denver Post: Drought Impacts to Irrigation July 19, 2013

Tempers flaring in Colorado over diminished irrigation water

By Nancy Lofholm

Posted: DenverPost.com

July 19, 2013

Sheldon Zwicker has been keeping an eye on irrigation ditches at his family's McElmo Canyon ranch since he was no taller than a cattail and was tagging along to help his grandfather water crops.

The 68-year-old remembers farmers and ranchers getting hot under the collar about water disputes then. And he's not surprised that, in the current drought year, the same squabbles have water users in ditchside near-brawls, threatening one another with shovels, pitchforks, guns and words.

Water has been fought for since before King Hammurabi first declared in his famous Code somewhere around 1795 B.C. that farmers needed to be neighborly about using it.

That's never been an easy code to follow throughout history because water is a paycheck to those in the agricultural trade. It's an especially tough sell these days in Colorado when all but a sliver of the state is in various levels of drought.

In the southwest corner of Colorado, where a dozen water users operating on a sort of honor system might rely on one ditch, the problem seems to be particularly bad.

When those users are tempted to take more than their share to sprinkle on crispy crops and slake the thirst of baking bovines, the number of disputes has gone up as fast as the midsummer thermometer.

"There's probably more disputes over water down here now than there is over wives. It's been a real trying year," Zwicker said. "When it's 110 (degrees) out and you're trying to get your crops wet, and you're out of water, and you find your neighbor has it, well, you blow your stack."

There are no statewide figures for the anecdotal spike in water disputes this year. Each county and the dozens of water districts and ditch companies in Colorado handle the trouble in different ways.

Montezuma County has had enough trouble that Sheriff Dennis Spruell has designated a water deputy to act as a mediator and an enforcer when neighbors call in accusing one another of stealing, hogging or wasting water.

That deputy has been fielding 20 to 30 calls a day lately, compared with four or five calls in the past.

He's not the only one overburdened by water woes. On some hot mornings, the Montezuma Valley Irrigation Co.'s water commissioner in Cortez has received 20 to 30 calls by coffee-break time.

Disputes range from finding suspicious shovel-built ditches to catching someone red-handed diverting water. One method of stealing water is for a water thief to open a gate in the middle of the night to divert water to his crops and to sneak back before daylight to reset the water flow and try to cover up the crime.

Drought maps can bring some understanding of this kind of desperation.

Montezuma County shows up on the drought map covered in orange — the color denoting extreme drought. In southeast Colorado, the fire-engine red identifying a worse rating — exceptional drought — has crept across more than 10 counties of farm country. Even with recent rains, Colorado is colored with the shades of lesser drought — the yellow of "severe drought" and the cream-colored tone of moderate drought. One bit of green now snakes down north central Colorado, which means an upgrade from drought to "abnormally dry."

In the "exceptional drought" area of southeast Colorado, the irrigation water situation has gotten so bad for a second year in a row that water squabbles are actually few between farmers who are all in the same dire boat.

"There isn't much dispute here over water because there isn't any," Fort Lyon Canal Co. manager Wesley Eck said.

Nearly three-fourths of the 93,000 acres the canal company supplies are without water this year because a depleted Arkansas River and its tributaries can no longer deliver. Farmers didn't even bother planting many of their crops this year.

In the southwest corner, the McPhee Reservoir is trickling 25 percent of water to farmers and ranchers, the Ute Mountain Ute tribe and a fishery.

"We not only have a limited supply, but the duration of the season will be cut short," said Mike Preston, general manager of the Dolores Water Conservancy District that disburses McPhee water.

That district uses pressured meters to track exactly how much water users get so there is less room for arguing about fair shares. Arguments tend to happen more when users have a direct flow of water in a canal that is dispersed to individual users' fields through the use of flow-controlling headgates.

"So everyone is kind of suffering together," Preston said.

Water woes have resulted in more disputes than just neighbors throwing punches and shovels. Legal actions are also in the arsenal with larger users.

In Craig, the U.S. Department of Justice has filed charges against two brothers who allegedly cut unauthorized ditches to take water from BLM lands to their ranches.

In the Canyons of the Ancients, in southwest Colorado, a citizens group has sued the BLM alleging the government is taking water that should belong to individual citizens.

The city of Ouray has asked a water court to change its water rights so agricultural users downstream can't cut back on the city's allotment in dry times.

And for farmers who aren't currently fighting about water, Nate Midcap, the manager of the Central Yuma Groundwater Management District in the relatively wet drought-stricken northeast corner of Colorado, has a prediction, given the fact that aquifers are dropping and heat waves are increasing: "Sooner or later they will be."

Nancy Lofholm: 970-256-1957, nlofholm@denverpost.com or twitter.com/nlofholm